

## 2 Basic Structures: Sets, Functions, Sequences, Sums, and Matrices

### Introduction

Chapter 2 of the textbook covers mathematical objects that are essential to the study of discrete mathematics. We will see how Maple represents these objects and how they correspond to the fundamental data structures used in Maple.

In Sections 1 and 2, we will see that Maple's implementation of a set corresponds naturally to the mathematical concept. We will also see how to represent fuzzy sets. In Section 3, we will consider three distinct ways in which the concept of function can be represented in Maple and how these three approaches can be used in different circumstances. Section 4 will look at the idea of sequence as it is used in Maple, which is somewhat different from the mathematical meaning of sequence, and how Maple can be used to compute both finite and symbolic summations. In Section 5, we will use Maple to list positive rational numbers in a way that demonstrates the fact that the rationals are enumerable. And in Section 6, we will see how Maple can be used to study matrices.

### 2.1 Sets

Sets are essential in the description of almost all of the discrete objects that we will study. They are also fundamental to Maple. As such, Maple provides extensive support for both their representation and manipulation.

#### *Set Basics*

As is standard in mathematics, you can create a set in Maple using the roster method by listing the elements of the set separated by commas and enclosed in braces. The elements of the set can be any of the objects known to Maple. Typical examples are shown here.

$$\begin{array}{ll} > \{1, 2, 3\} \\ & \{1, 2, 3\} \end{array} \quad (2.1)$$

$$\begin{array}{ll} > \{“a”, “b”, “c”\} \\ & \{“a”, “b”, “c”\} \end{array} \quad (2.2)$$

$$\begin{array}{ll} > \{\{1, 2\}, \{1, 3\}, \{2, 3\}\} \\ & \{\{1, 2\}, \{1, 3\}, \{2, 3\}\} \end{array} \quad (2.3)$$

$$\begin{array}{ll} > \{\} \\ & \emptyset \end{array} \quad (2.4)$$

$$\begin{array}{ll} > \{[1, 2], [2, 5], [3, 11]\} \\ & \{[1, 2], [2, 5], [3, 11]\} \end{array} \quad (2.5)$$

In the first two examples above, the sets contain the numbers 1, 2, and 3, and the characters  $a$ ,  $b$ , and  $c$ , respectively. In the third example, the elements of the set are themselves sets. The fourth

example is the empty set. And in the final example, the elements of the set are 2-element lists, which we can interpret as a set containing three ordered pairs.

Note that Maple's idea of a set corresponds to the mathematical notion. In particular, there is no notion of "multiplicity" for set members, nor is the order of elements relevant. For example, consider the sets defined below.

```
> set1 := {1,2,3,1}
    set1 := {1,2,3} (2.6)
```

Observe that the duplicates in the input above are only included once in the set.

```
> set2 := {2,3,1}
    set2 := {1,2,3} (2.7)
```

The second example illustrates that order is irrelevant for sets. Maple sorted the input into numerical order in order to improve efficiency in computations (e.g., if we have a large set that we want to search to see if a particular element is in the set or not, having the elements of the set in a particular order can make that search run much more quickly). Maple puts the elements of a set into a canonical order, but it understands that the order is irrelevant from a mathematical perspective. If order is important in a particular context or if repeated elements are allowed, you should use a list instead of a set.

To confirm that two sets  $A$  and  $B$  are equal, we use the **evalb** (evaluate Boolean) command on the proposition  $A = B$ .

```
> evalb(set1 = set2)
    true (2.8)
```

### Selection

A useful consequence of the fact that Maple stores sets in a particular order is that you can use the selection operator to access individual elements. To select an individual element from a set, you enclose the index of the element in brackets, as below.

```
> set3 := {"a","b","c","d","e","f"}
    set3 := {"a","b","c","d","e","f"} (2.9)
```

```
> set3[2]
    "b" (2.10)
```

Negative values count from the right, so the second to last entry (according to Maple's imposed order) is accessed as follows.

```
> set3[-2]
    "e" (2.11)
```

Multiple entries can be accessed by using a range instead of a single value.

```
> set3[3..5]
    {"c","d","e"} (2.12)
```

By putting a list within the selection brackets, it's possible to obtain any subset you wish. Note that double brackets are required as the outer set of brackets is representing the selection operator and the inner set of brackets is enclosing the list of indices being accessed.

```
> set3[[1,3,5]]
      {"a","c","e"}
(2.13)
```

### The seq Command

One of the most useful commands for constructing sets or lists is the **seq** (for sequence) command. For example, we can use **seq** to build the set consisting of the squares of some integers. The **seq** command requires two arguments. The first argument in this case will be the expression  $i^2$ , which indicates that the elements of the sequence will be the square of the value of the index variable **i**. The second argument will be **i=-10..10** which indicates that the index variable **i** will range from -10 to 10.

```
> seq1 := seq(i^2,i=-10..10)
      seq1 := 100, 81, 64, 49, 36, 25, 16, 9, 4, 1, 0, 1, 4, 9, 16, 25,
              36, 49, 64, 81, 100
(2.14)
```

Note that the **seq** command has produced a sequence of values and that the values are listed in the order in which they were generated and with repetition. We make a set from these values by enclosing the sequence in braces; we also create the set of fourth powers of integers between -10 and 10.

```
> set4 := {seq1}
      set4 := {0, 1, 4, 9, 16, 25, 36, 49, 64, 81, 100}
(2.15)
```

```
> set5 := {seq(i^4,i=-10..10)}
      set5 := {0, 1, 16, 81, 256, 625, 1296, 2401, 4096, 6561, 10000}
(2.16)
```

### Membership, Subset, and Size

Perhaps the most basic question one can ask about a set is whether or not a particular object is or is not a member of a set. In Maple, you do this with the **in** operator. To check that  $4 \in \text{set4}$  but  $5 \notin \text{set4}$  we enter the following.

```
> evalb(4 in set4)
      true
(2.17)
```

```
> evalb(5 in set4)
      false
(2.18)
```

Note that we need to use **evalb** so that Maple will compute the truth value. Without it, Maple will just restate the proposition, as below.

```
> 6 in set4
      6 ∈ {0, 1, 4, 9, 16, 25, 36, 49, 64, 81, 100}
(2.19)
```

As an alternative to the **in** operator, we can use the **member** command. This command does not require the application of **evalb** to return a Boolean result, but is called as a function applied to

arguments rather than the infix operator form. The first argument of **member** is the object and the second is a set or list or other larger structure.

```
> member(4, set4)
true
```

(2.20)

One advantage of **member** is that it accepts an unevaluated name as an optional third argument. Upon execution, the location of the object, or first location if it appears more than once, will be stored in the name, provided it is present. If the object is not present, no assignment is made to the name.

```
> member(36, set4, 'locationof36')
true
```

(2.21)

```
> locationof36
7
```

(2.22)

Note the use of single quotes around **locationof36** in the call to **member**. This prevents the name being evaluated to any value that may have been previously assigned to it and ensures that the function will execute even if the name has previously been used. The single quotes are not required, and omitting them can be useful if you want to avoid overwriting data.

The inclusion in a parameter sequence of an optional unevaluated name is common in Maple functions, and it is worth briefly illustrating one reason it is a useful feature. Suppose we are writing a procedure that needs to check whether a set (or list or other appropriate data structure) contains an object and then take action on that member. For example, we may wish to replace 81 with 18 in **set4**. We can take advantage of the fact that **member** has the side-effect of recording the location to do this.

```
> if member(81, set4, 'locationof81') then
    set4 := subsop(locationof81 = 18, set4)
else
    print("Not found")
end if;
set4 := {0, 1, 4, 9, 16, 18, 25, 36, 49, 64, 100}
```

(2.23)

Recall that the first argument of **subsop** is an equation with a position on the left-hand side and a new value on the right-hand side. Also recall that it does not modify the data structure given as the second argument, so we reassign the list in order to update it.

Maple provides the **subset** operator to check whether or not one set is a subset of another. For example,

```
> {1, 2} subset {1, 2, 3}
true
```

(2.24)

```
> {} subset {1, 2, 3}
true
```

(2.25)

```
> {1, 2, 5} subset {1, 2, 3}
false
```

(2.26)

In the previous chapter, we made use of the **nops** command to determine the number of elements in a list. This command also calculates the size of a finite set.

```
> nops(set5)
11
```

(2.27)

Another way to find the size of a finite set or the length of a list is the **numelems** command. For a list or a set, **numelems** and **nops** return the same result.

```
> numelems(set5)
11
```

(2.28)

Also note that both of them return the number of top-level elements. For example, on each list of lists below, they give the number of sublists, not the total number of objects. This is consistent with the mathematical definition of the size of a set.

```
> numelems([[1, 2, 3, 4, 5], [6, 7, 8]])
2
```

(2.29)

```
> nops([[1, 2, 3], [4, 5, 6], [7, 8, 9], [10, 11, 12]])
4
```

(2.30)

However, **numelems** is more flexible; for example, returning the number of characters in a string, which **nops** would treat as a single entity.

```
> numelems("helloworld")
11
```

(2.31)

```
> nops("helloworld")
1
```

(2.32)

The **numelems** command will also delve down into data structures such as matrices to report the number of entries, whereas **nops** would return the number of rows. In this way, **numelems** generally produces the more natural output for the context, while **nops** is focused on the formal representation of objects.

## Power Sets

Maple has a built-in command to compute the power set of a finite set. The **powerset** command is part of the **combinat** combinatorics package and accepts a set as an argument and returns the power set of the given set.

We saw in the first chapter that commands that are part of Maple packages can be used in one of two ways. The long form consists of the name of the package followed by the name of the command in brackets and then the arguments in parentheses.

```
> combinat[powerset]({1, 2, 3})
{∅, {1}, {2}, {3}, {1, 2}, {1, 3}, {2, 3}, {1, 2, 3}}
```

(2.33)

In order to use the short form of the calling sequence, that is, to be able to omit the name of the package, you must first use the **with** command to load the package.

```
> with(combinat)
[Chi, bell, binomial, cartprod, character, choose, composition, conjpart,
 decodepart, encodepart, eulerian1, eulerian2, fibonacci, firstcomb,
 firstpart, firstperm, graycode, inttovec, lastcomb, lastpart, lastperm,
 multinomial, nextcomb, nextpart, nextperm, numbcomb, numbcomp,
 numbpert, numbperm, partition, permute, powerset, prevcomb,
 prevpart, prevperm, randcomb, randpart, randperm, rankcomb,
 rankperm, setpartition, stirling1, stirling2, subsets,
 unrankcomb, unrankperm, vectoint]
```

(2.34)

The result of this command is to list the commands that have been made available. Typically, you will end a **with** statement with a colon to suppress this output. Note that you can selectively load only those commands from a package that you actually need by following the name of the package with the names of the commands separated by commas.

```
> with(combinat, powerset, subsets, cartprod)
[powerset, subsets, cartprod]
```

(2.35)

However you choose to load the command, this makes it possible to call the command without the name of the package.

```
> powerset({"a", "b"})
{{}, {"a"}, {"b"}, {"a", "b"}}
```

(2.36)

Keep in mind that it is often best to use the long form names in procedures you write, so that the procedure is not dependent on the package having been loaded.

### The subsets Command

The textbook mentions that the size of the power set of a set is  $2^n$  where  $n$  is the size of the original set. For even reasonably sized sets, their power set can be very large and can easily tax your computer's memory. For this reason, Maple provides a second command for computing with power sets called **subsets**, which is also in the **combinat** package. The result of applying **subsets** is not a list of sets. Instead, the **subsets** command returns a table. Tables will be discussed in detail in Section 3 of this chapter; for now, it is enough to see how to use the **subsets** command.

Consider the set consisting of the first ten positive integers.

```
> firstTen := {seq(1..10)}
firstTen := {1, 2, 3, 4, 5, 6, 7, 8, 9, 10}
```

(2.37)

Note the alternate form of the **seq** command, which is equivalent to **seq(i,i=1..10)**.

Apply the **subsets** command and store the result as **subsetTen**.

```
> subsetTen := subsets(firstTen):
```

It is typical to suppress the output as we have done here. Those readers who are curious can issue the command with the output displayed, but do not expect to see any subsets listed in the output from **subsets**.

As mentioned, the result of the **subsets** command is a table containing two objects. One of these objects is a procedure, which is executed with the syntax **subsetTen[nextvalue]()**; The first time this procedure is called, it will return the empty set.

```
> subsetTen[nextvalue]()
 $\emptyset$  (2.38)
```

Each subsequent time it is called, it returns the “next” subset of the given set.

```
> subsetTen[nextvalue]()
{1} (2.39)
```

```
> subsetTen[nextvalue]()
{2} (2.40)
```

```
> subsetTen[nextvalue]()
{3} (2.41)
```

```
> subsetTen[nextvalue]()
{4} (2.42)
```

The other object in the **subsetTen** table is a Boolean value named **finished**. This value is accessed by

```
> subsetTen[finished]
false (2.43)
```

Once the **subsetTen[nextvalue]** procedure has returned the “last” subset, which will always be the original set itself, this Boolean is set to true. This provides a way to control a while loop. By setting the condition in the while loop to **not subsetTen[finished]**, the loop continues until all of the subsets have been considered.

### Example Using the subsets Command

As an example of a practical use of this command, we will search for the subsets of the first five positive integers which have their own cardinality as a member, that is, those sets  $S$  such that  $|S| \in S$ . We do this by considering each subset in turn and checking whether its size, obtained using the **nops** command, is **in** the set.

Here is the procedure that will list all subsets of the first five positive integers whose cardinalities are members of themselves..

```
1 selfSize := proc ()
2   local onetofive, pSet, S, n;
3   onetofive := {seq(1..5)};
4   pSet := combinat[subsets](onetofive);
5   while not pSet[finished] do
```

```

6      S := pSet[nextvalue]();
7      if nops(S) in S then
8          print(S);
9      end if;
10     end do;
11 end proc:

```

After declaring local variables, we form the set of the integers from 1 to 5. Then, we apply the **subsets** command to form **pSet**. Remember that this is not actually the power set, it is the table described above. Then, we begin a while loop, which continues until **pSet[finished]** is true. Remember that **pSet[finished]** is false until the **pSet[nextvalue]** procedure produces the final subset. Inside the while loop, we print those sets which have their own size as a member.

Now, execute the procedure.

```

> selfSize()
{1}
{1,2}
{2,3}
{2,4}
{2,5}
{1,2,3}
{1,3,4}
{1,3,5}
{2,3,4}
{2,3,5}
{3,4,5}
{1,2,3,4}
{1,2,4,5}
{1,3,4,5}
{2,3,4,5}
{1,2,3,4,5}

```

(2.44)

It is worth reiterating the purpose of the **subsets** command in contrast with the **powerset** command. For sets of any significant size, calculating the power set can be very taxing both on a computer's memory and with regards to time. In an example like **selfSize**, the **subsets** command avoids the need to store the entire powerset by generating and testing one subset at a time. Additionally, suppose that instead of listing all of the subsets with a given property, we only wanted to find an example of a set with that property, for instance, to find a counterexample. In those circumstances, we could return the example as soon as it was found and save the time it would have taken **powerset** to have calculated all of the subsets.

### **Cartesian Product**

The **combinat** package also has a command for calculating the Cartesian product of sets called **cartprod**. This command is very similar to the **subsets** command except that the



**nextvalue** procedure returns a list representing the “next” element in the Cartesian product of the given sets.

As a first example, recall that Example 17 from Section 2.1 of the text computes the Cartesian product of  $\{1, 2\}$  and  $\{a, b, c\}$  to be  $\{(1, a), (1, b), (1, c), (2, a), (2, b), (2, c)\}$ . To compute this product in Maple, we use the **cartprod** function. It accepts only one argument: a list whose members are the sets whose product is desired. We store the result as **Ex17** and, as before, we suppress the output for the result.

```
> Ex17 := cartprod([ {1, 2}, {“a”, “b”, “c”} ]) :
```

We can now display the elements one at a time using the same kind of loop as in the **selfSize** procedure. We create a while loop that continues as long as **Ex17[finished]** is false and prints the result of **Ex17[nextvalue]()**.

```
> while not Ex17[finished] do
    print(Ex17[nextvalue]())
end do;
[1, “a”]
[1, “b”]
[1, “c”]
[2, “a”]
[2, “b”]
[2, “c”]
(2.45)
```

Maple displays the entries in the Cartesian product as two-element lists.

The argument to **cartprod** can be a list of any number of sets. To compute the members of the Cartesian product of three sets, we include three sets in the list. Note that this command accepts either sets or lists as the objects to be multiplied. Below, we expand our previous example and compute  $\{1, 2\} \times \{a, b, c\} \times \{\pi, e\}$ . We give the last set as a list instead of a set to illustrate the command’s flexibility. (Note that we obtain  $\pi$  with the Maple constant **Pi**, and  $e$  is obtained by applying the exponential function **exp** with exponent 1.)

```
> cartesian3 := cartprod([ {1, 2}, {“a”, “b”, “c”}, [Pi, e] ]) :

> while not cartesian3[finished] do
    print(cartesian3[nextvalue]())
end do;
[1, “a”, Pi]
[1, “a”, e]
[1, “b”, Pi]
[1, “b”, e]
[1, “c”, Pi]
[1, “c”, e]
[2, “a”, Pi]
[2, “a”, e]
[2, “b”, Pi]
```

$$\begin{aligned}
& [2, "b", e] \\
& [2, "c", \pi] \\
& [2, "c", e]
\end{aligned}
\tag{2.46}$$

Maple does not include a command analogous to **powerset** for computing the Cartesian product all at once as a single set of elements. The task of creating such a command is left to the reader.

## 2.2 Set Operations

In this section, we will examine the commands Maple provides for computing set operations. Then, we will use these commands and the concept of membership tables to see how Maple can be used to prove set identities. Finally, we see how we can use Maple to represent and manipulate fuzzy sets.

### Basic Operations

Maple provides fairly intuitive commands related to the basic set operations of union, intersection, and set difference. The commands are named **union**, **intersection**, and **minus** and, like the logical connectives from the previous chapter, are infix operators. For example, consider the following sets.

$$\begin{aligned}
& > \text{primes} := \{2, 3, 5, 7, 11, 13\} \\
& \text{primes} := \{2, 3, 5, 7, 11, 13\}
\end{aligned}
\tag{2.47}$$

$$\begin{aligned}
& > \text{odds} := \{1, 3, 5, 7, 9, 11, 13\} \\
& \text{odds} := \{1, 3, 5, 7, 9, 11, 13\}
\end{aligned}
\tag{2.48}$$

We compute their union and intersection as follows.

$$\begin{aligned}
& > \text{primes} \text{ union } \text{odds} \\
& \{1, 2, 3, 5, 7, 9, 11, 13\}
\end{aligned}
\tag{2.49}$$

$$\begin{aligned}
& > \text{primes} \text{ intersection } \text{odds} \\
& \{3, 5, 7, 11, 13\}
\end{aligned}
\tag{2.50}$$

The set difference is obtained by use of the **minus** operator. The following examples illustrate the fact that, unlike, union and intersection, set difference is not symmetric, that is,  $A - B$  is generally not the same as  $B - A$ .

$$\begin{aligned}
& > \text{primes} \text{ minus } \text{odds} \\
& \{2\}
\end{aligned}
\tag{2.51}$$

$$\begin{aligned}
& > \text{odds} \text{ minus } \text{primes} \\
& \{1, 9\}
\end{aligned}
\tag{2.52}$$

Maple does not provide a complement command. Such a command would be ambiguous with regards to the universe that should be applied. Instead, you must compute complements with the minus operator. For example, the complement of the **primes** set in the universe consisting of the positive integers from 1 to 13 is computed as follows.

$$\begin{aligned}
& > \text{universe} := \{\text{seq}(1..13)\} \\
& \text{universe} := \{1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13\}
\end{aligned}
\tag{2.53}$$

> *universe minus primes*  
 $\{1, 4, 6, 8, 9, 10, 12\}$  (2.54)

The **union** operator is often used to build sets within procedures. As an example, consider the following procedure that creates the set of the squares of the first ten positive integers.

```

1  tenSquares := proc ()
2      local S, i;
3      S := {};
4      for i from 1 to 10 do
5          S := S union {i^2};
6      end do;
7      return S;
8  end proc;
```

> *squares := tenSquares()*  
 $squares := \{1, 4, 9, 16, 25, 36, 49, 64, 81, 100\}$  (2.55)

After the declaration of local variables, we initialize **S**, the set being built, to the empty set. Inside the for loop, we produce the square of the loop index and add it to the set **S** by setting **S** equal to its union with the singleton containing the square of the index variable. While this is a very simple example, and requires much more typing than the **seq** command would have, it illustrates a common technique.

All three operators can also be used as commands, and, in the cases of **union** and **intersect**, can take any number of sets as arguments. For example, to compute the union of three sets, you can use either of the two methods below.

> *primes union odds union squares*  
 $\{1, 2, 3, 4, 5, 7, 9, 11, 13, 16, 25, 36, 49, 64, 81, 100\}$  (2.56)

> 'union'(primes, odds, squares);  
 $\{1, 2, 3, 4, 5, 7, 9, 11, 13, 16, 25, 36, 49, 64, 81, 100\}$  (2.57)

Note that the single left quotes are required in the second option, since it is in 1-D notation. Generally, single left quotes are used to tell Maple that the string of characters they enclose is a name. In the first of the statements above, the **union** keyword is used to invoke the operator. In the second, a procedure whose name is **union** is executed. The single left quotes are required in order to use a keyword as a name. In 2-D notation, they are not required.

### ***Set Identities and Membership Tables***

The textbook discusses how membership tables can be used to prove set identities. We will use the idea of membership tables to have Maple prove set identities.

A membership table is very similar to a truth table. In a membership table, each row corresponds to a possible element in the universe. We use 1 and 0 to indicate that the element corresponding to that row is or is not in the set.

### **An Illustration of the Approach with an Example**

Let us look at a specific example in detail in order to get an idea of how we can use Maple to automate the construction of membership tables. Consider the De Morgan's law  $\overline{A \cup B} = \overline{A} \cap \overline{B}$ .

We begin the table by considering all possible combinations of 1s and 0s for  $A$  and  $B$  and add columns for the two sides of the identity. (Ordinarily, when doing this by hand, you would add columns for the intermediary steps as well.)

Row number	$A$	$B$	$\overline{A \cup B}$	$\overline{A} \cap \overline{B}$
1	1	1		
2	1	0		
3	0	1		
4	0	0		

Row number

$A$

$B$

$\overline{A \cup B}$

$\overline{A} \cap \overline{B}$

1

1

1

2

1

0

3

0

1

4

0

0

We can determine the values for the last two columns as follows. Let the universe be the set consisting of the row numbers  $\{1, 2, 3, 4\}$ . Now, form sets  $A$  and  $B$  as follows: a value in the universe of row numbers is in  $A$  if there is a 1 in the column for  $A$  in that row. Thus  $A = \{1, 2\}$  because rows 1 and 2 have 1s in the column for  $A$ . Likewise,  $B$  is defined to be  $\{1, 3\}$ .

$$\begin{aligned} &> \text{rows} := \{1, 2, 3, 4\} \\ &\quad \text{rows} := \{1, 2, 3, 4\} \end{aligned} \tag{2.58}$$

$$\begin{aligned} &> \text{setA} := \{1, 2\} \\ &\quad \text{setA} := \{1, 2\} \end{aligned} \tag{2.59}$$

$$\begin{aligned} &> \text{setB} := \{1, 3\} \\ &\quad \text{setB} := \{1, 3\} \end{aligned} \tag{2.60}$$

Next, compute both sides of the identity  $\overline{A \cup B} = \overline{A} \cap \overline{B}$ . Remember that we must use set differences to obtain the complements.

$$\begin{aligned} &> \text{rows minus (setA union setB)} \\ &\quad \{4\} \end{aligned} \tag{2.61}$$

This indicates that row 4 is the only row with a 1 in the column for  $\overline{A \cup B}$ .

$$\begin{aligned} &> (\text{rows minus setA}) \text{ intersect } (\text{rows minus setB}) \\ &\quad \{4\} \end{aligned} \tag{2.62}$$

This tells us that row 4 is also the only row with a 1 in the column for  $\overline{A} \cap \overline{B}$ . Since the two sets are equal, the two columns must be identical.

The above indicates the approach that we will be using. First, compute the initial entries in the rows of the membership table; each row corresponds to a different assignment of 1s and 0s. Second, construct sets whose entries are the row numbers corresponding to 1s in the table. And finally, compute both sides of the identity. If the resulting sets are equal, then we have confirmed the identity.

### Revising the GetVars Procedure

Much of what we do here will be very similar to how we created the **AreEquivalent** procedure in Section 1.3 of this manual. First, we create expressions representing the two sides of the identity from Example 14 of the text. We use the name **U** for the universe.

$$\begin{aligned} &> \text{Ex14L} := U \text{ minus } (A \text{ union } (B \text{ intersect } C)) \\ &\quad \text{Ex14L} := U \setminus (A \cup B \cap C) \end{aligned} \quad (2.63)$$

$$\begin{aligned} &> \text{Ex14R} := ((U \text{ minus } C) \text{ union } (U \text{ minus } B)) \text{ intersect } (U \text{ minus } A) \\ &\quad \text{Ex14R} := (U \setminus A) \cap (U \setminus B \cup (U \setminus C)) \end{aligned} \quad (2.64)$$

$$\begin{aligned} &> \text{Ex14} := \text{Ex14L} = \text{Ex14R} \\ &\quad \text{Ex14} := U \setminus (A \cup B \cap C) = (U \setminus A) \cap (U \setminus B \cup (U \setminus C)) \end{aligned} \quad (2.65)$$

Now, we revive the **GetVars** procedure from Section 1.3.

```

1  GetVars := proc(exp)
2      local L, i, j;
3      L := [exp];
4      i := 1;
5      while i <= nops(L) do
6          if type(L[i], name) then
7              i := i + 1;
8          else
9              L := subsop(i=op(L[i]), L);
10             end if;
11         end do;
12         L := {op(L)} minus {U};
13         return [op(L)];
14     end proc;
```

$$\begin{aligned} &> \text{Ex14Vars} := \text{GetVars}(\text{Ex14}) \\ &\quad \text{Ex14Vars} := [A, B, C] \end{aligned} \quad (2.66)$$

Note that this is identical to the procedure we created in Section 1.3 with one small change. In the next to last line we convert **L** into a set and remove the name **U** from it. In these procedures, we will always consider **U** to be the name of the universe. The last line turns the set of variables back into a list. This is not necessary, strictly speaking, but it is more natural to consider the variables stored in a list, and therefore with order.

### Producing the Rows of the Table

In Section 1.3, we created a procedure called **nextTA**. This procedure was responsible for producing the truth value assignments for the variables. In other words, it produced the rows of the truth table. Look again at the membership table above. Observe that the rows correspond to the members of the Cartesian product

$$\{0, 1\} \times \{0, 1\} = \{(0, 0), (0, 1), (1, 0), (1, 1)\}.$$

$$\text{delayCrossProduct}(\{0, 1\}, \{0, 1\}) = \{0, 1\}.$$

In fact, the definition of the Cartesian product is exactly suited to what we need. The rows of the table are all the possible choices of 0s and 1s for the variables. The Cartesian product of  $\{0, 1\}$  with itself is the collection of all possible tuples with each entry in the tuple equal to 0 or to 1.

We can use the following code to produce the Cartesian product with 3 variables.

```
> CartesianMembership := cartprod([seq({0, 1}, i = 1 .. 3)]) colon
```

Note the use of **seq** to create three copies of the set  $\{0, 1\}$ .

```
> while not CartesianMembership[finished] do  
  CartesianMembership[nextvalue]()
```

```
end do :
```

```
[0, 0, 0]
```

```
[0, 0, 1]
```

```
[0, 1, 0]
```

```
[0, 1, 1]
```

```
[1, 0, 0]
```

```
[1, 0, 1]
```

```
[1, 1, 0]
```

```
[1, 1, 1]
```

(2.67)

You can see that the results are identical to the first three columns of Table 2 of Section 2.2 in the textbook, albeit in reverse order.

### Building Sets to Correspond to the Table Rows

We need to build sets whose entries are determined by the rows of the membership table (i.e., by the elements of a Cartesian product of  $\{0, 1\}$  as above). The sets, corresponding to what we called **setA** and **setB** in the example at the start of this subsection, will be stored in a list. That is, we will create a list of sets. These sets are identified with the variables in the identity to be checked as follows: the set in position  $i$  in the list of sets corresponds to the variable in position  $i$  in the list that results from **GetVars**.

Begin by initializing a list of the right size (the number of variables) whose entries are the empty set. We use the **seq** command again to create multiple copies.

```
> MTableSets := [seq({}, i = 1 .. 3)]
```

```
MTableSets := [∅, ∅, ∅]
```

(2.68)

Note that we can access and modify the lists as usual. For instance, to add 5 to the second set:

```
> MTableSets[2] := MTableSets[2] union {5}
MTableSets2 := {5} (2.69)
```

```
> MTableSets
[∅, {5}, ∅] (2.70)
```

We reinitialize this so we can use it below.

```
> MTableSets := [seq({}, i = 1..3)]
MTableSets := [∅, ∅, ∅] (2.71)
```

We reexecute the **cartprod** command from above in order to reset it. We also need an index variable that we initialize to 0.

```
> CartesianMembership := cartprod([seq({0, 1}, i = 1..3)]) :
> MTrownum := 0
MTrownum := 0 (2.72)
```

Now, we use the standard Cartesian product while loop, but instead of just calculating and displaying the 3-tuple, we use an inner loop to consider each entry (i.e., variable) in turn. If the value is 1, we add the current value of **MTrownum** to the corresponding set. In this example, we have the loop print out the index, the tuple from the Cartesian product, and the current state of **MTableSets**.

```
> while not CartesianMembership[finished] do
  MTrownum := MTrownum + 1 :
  currentTuple := CartesianMembership[nextvalue]() :
  for varI from 1 to 3 do
    if currentTuple[varI] = 1 then
      MTableSets[varI] := MTableSets[varI] union {MTrownum};
    end if :
  end do :
  print(MTrownum, currentTuple, MTableSets);
end do :
1, [0, 0, 0], [∅, ∅, ∅]
2, [0, 0, 1], [∅, ∅, {2}]
3, [0, 1, 0], [∅, {3}, {2}]
4, [0, 1, 1], [∅, {3, 4}, {2, 4}]
5, [1, 0, 0], [{5}, {3, 4}, {2, 4}]
6, [1, 0, 1], [{5, 6}, {3, 4}, {2, 4, 6}]
7, [1, 1, 0], [{5, 6, 7}, {3, 4, 7}, {2, 4, 6}]
8, [1, 1, 1], [{5, 6, 7, 8}, {3, 4, 7, 8}, {2, 4, 6, 8}] (2.73)
```

We also need the universe represented.

```
> Ex14U := {seq(1..MTrownum)}
Ex14U := {1, 2, 3, 4, 5, 6, 7, 8} (2.74)
```



## Evaluating the Identity for Each Row

Once the list of sets is built up, all that remains is to evaluate the identity with these sets in place of the names. We do this with the **eval** and **zip** commands (recall that we previously used this technique in Section 1.3 of this manual).

First, we use **zip** to create equations that identify the variables with the corresponding sets.

```
> Ex14eqns := zip((a,b) → a = b, Ex14Vars, MTableSets)
Ex14eqns := [A = {5,6,7,8}, B = {3,4,7,8}, C = {2,4,6,8}] (2.75)
```

We add the equation **U=Ex14U**.

```
> Ex14eqns := [op(Ex14eqns), U = Ex14U]
Ex14eqns := [A = {5,6,7,8}, B = {3,4,7,8}, C = {2,4,6,8},
U = {1,2,3,4,5,6,7,8}] (2.76)
```

And then we apply **eval** to perform the substitution and apply **evalb** to obtain a truth value.

```
> eval(Ex14, Ex14eqns)
Ex14L = {1,2,3} (2.77)
```

```
> evalb((2.77))
false (2.78)
```

## The Procedure

Finally, we combine it all into a single procedure.

```
1 MemberTable := proc(identity)
2   local vars, numvars, cartesian, setList, rowNum, curTuple, vI,
   Universe, eqns, U;
3   vars := GetVars(identity);
4   numvars := nops(vars);
5   cartesian := combinat[cartprod]([seq({0,1}, i=1..numvars)]);
6   setList := [seq({}, i=1..numvars)];
7   rowNum := 0;
8   while not cartesian[finished] do
9     rowNum := rowNum + 1;
10    curTuple := cartesian[nextvalue]();
11    for vI from 1 to numvars do
12      if curTuple[vI] = 1 then
13        setList[vI] := setList[vI] union {rowNum};
14      end if;
15    end do;
16  end do;
17  Universe := {seq(i, i=1..rowNum)};
18  eqns := zip((a,b) → a=b, vars, setList);
19  eqns := [op(eqns), U=Universe];
20  return evalb(eval(identity, eqns));
21 end proc;
```

We can now demonstrate:  $(A - B) - C = (A - C) - (B - C)$ .

$$\begin{aligned} &> \text{MemberTable}((A \text{ minus } B) \text{ minus } C = (A \text{ minus } C) \text{ minus } (B \text{ minus } C)) \\ &\quad \text{true} \end{aligned} \tag{2.79}$$

However,  $\overline{A \cup B} \neq \overline{A} \cup \overline{B}$ .

$$\begin{aligned} &> \text{MemberTable}(U \text{ minus } (A \text{ union } B) = (U \text{ minus } A) \text{ union } (U \text{ minus } B)) \\ &\quad \text{false} \end{aligned} \tag{2.80}$$

## Computer Representation of Fuzzy Sets

The textbook describes a way to represent sets as bit strings in order to efficiently store and compute with them. Here, we will explore this idea further in order to see how we can represent fuzzy sets in Maple. Fuzzy sets are described in the preamble to Exercise 73 in Section 2.2.

### Two Representations of Fuzzy Sets

In a fuzzy set, every element has an associated degree of membership, which is a real number between 0 and 1. We will represent fuzzy sets in three different ways.

The first way we can represent a fuzzy set in Maple is to combine the element with the degree of membership as a two-element list. For example, if the elements of our fuzzy set are the letters “a,” “b,” and “e,” where “a” has degree of membership 0.3, “b” has degree 0.7, and “e” has degree 0.1, then we would represent the set as:

$$\begin{aligned} &> \text{fuzzyR} := \{["a", 0.3], ["b", 0.7], ["e", 0.1]\} \\ &\quad \text{fuzzyR} := \{["a", 0.3], ["b", 0.7], ["e", 0.1]\} \end{aligned} \tag{2.81}$$

We refer to this as the “roster representation.”

The second approach is to use a “fuzzy-bit string” in essentially the same way as described in the text. First, we need to specify the universe and impose an order on it. Suppose the universe consists of the letters “a” through “g” ordered alphabetically. We represent the universe in Maple as a list so that the order we impose is preserved.

$$\begin{aligned} &> \text{fuzzyU} := ["a", "b", "c", "d", "e", "f", "g"] \\ &\quad \text{fuzzyU} := ["a", "b", "c", "d", "e", "f", "g"] \end{aligned} \tag{2.82}$$

Then, the fuzzy-bit string for the set **fuzzyR** will be the list of the degrees of membership of each element of the universe with 0 indicating nonmembership.

$$\begin{aligned} &> \text{fuzzyBitS} := [0.3, 0.7, 0, 0, 0.1, 0, 0] \\ &\quad \text{fuzzyBitS} := [0.3, 0.7, 0, 0, 0.1, 0, 0] \end{aligned} \tag{2.83}$$

The third approach is to make use of Maple’s built-in support for multisets. Recall from the textbook that in a multiset, objects can appear more than once. In a Maple **MultiSet**, objects are allowed to have nonnegative integers as their multiplicities. To represent fuzzy sets, we need to use the generalized version of the **MultiSet**, which allows the multiplicities to be any real numbers. The reason for the distinction is that for nongeneralized multisets, it is possible to iterate over the members of the set, which is not possible when the degrees of membership are not nonnegative integers.

A generalized **MultiSet** is created by applying the command to a specification of the elements of the set. One of the ways to specify the membership is the same as the roster representation above.

$$\begin{aligned} &> \text{fuzzyM} := \text{MultiSet}[\text{generalized}] ([\text{"a"}, 0.3], [\text{"b"}, 0.7], [\text{"e"}, 0.1]) \\ &\quad \text{fuzzyM} := \{[\text{"a"}, 0.3], [\text{"b"}, 0.7], [\text{"e"}, 0.1]\} \end{aligned} \quad (2.84)$$

If you wish, instead of lists containing the object and its degree of membership, you can specify the elements by equations, as illustrated below.

$$\begin{aligned} &> \text{MultiSet}[\text{generalized}] (\text{"a"} = 0.3, \text{"b"} = 0.7, \text{"e"} = 0.1) \\ &\quad \{[\text{"a"}, 0.3], [\text{"b"}, 0.7], [\text{"e"}, 0.1]\} \end{aligned} \quad (2.85)$$

Note that the display of a **MultiSet** is identical to the roster representation, but they are in fact different objects. Our roster representation is a set, while this representation is not.

$$\begin{aligned} &> \text{type}(\text{fuzzyM}, \text{set}) \\ &\quad \text{false} \end{aligned} \quad (2.86)$$

The usual set operations, **union**, **intersect**, and **minus**, can be applied to **MultiSet** objects. When applied to representations of fuzzy sets, **union** and **intersect** produce the expected result. However, because a generalized **MultiSet** is allowed to have negative values for its degree of membership, is defined to simply subtract the respective degrees of membership and can result in sets that are not fuzzy sets.

### Converting from Bit String to Roster Representation

Converting from a fuzzy-bit string to the roster representation is fairly straightforward. Use a for loop with index running from 1 to the number of elements in the universe. For each index, if the entry in the fuzzy-bit string is nonzero, then we add to the roster the pair consisting of the element from the universe and the degree of membership.

```

1 BitToRoster := proc (bitstring, universe)
2   local S, i;
3   S := {};
4   for i from 1 to nops(universe) do
5     if bitstring[i] <> 0 then
6       S := S union { [universe[i], bitstring[i]] };
7     end if;
8   end do;
9   return S;
10 end proc;
```

$$\begin{aligned} &> \text{BitToRoster}(\text{fuzzyBitS}, \text{fuzzyU}) \\ &\quad \{[\text{"a"}, 0.3], [\text{"b"}, 0.7], [\text{"e"}, 0.1]\} \end{aligned} \quad (2.87)$$

### Converting from Roster to Bit String Representation

In the other direction, we initialize a bit string to the 0-string. Then, we consider each member of the roster representation in turn, using the **for object in set do** form of a for loop. For every member of the set, we will need to determine the position of the set member in the universe in order to change the correct bit in the fuzzy-bit string.

To do this, we make use of the **member** command. Like the **in** operator, **member** will return true or false depending on whether or not the first argument is a member of the list (or set) given as the second argument. For example,

```
> member(3, {1, 2, 3, 4, 5})
true
```

(2.88)

```
> member(7, {1, 2, 3, 4, 5})
false
```

(2.89)

As we mentioned above, **member** also accepts a third, optional, argument which must be an unevaluated name. If the object is in fact a member of the set or list, then the position of the object is assigned to the name. (If the object is repeated, the location of the first occurrence is stored in the name.) We surround the name with right single quotes to prevent evaluation. Without the single quotes, the name could evaluate to a value that was previously assigned and Maple would not be able to assign to the name.

```
> member("d", ["a", "f", "s", "g", "d", "q"], 'pos')
true
```

(2.90)

```
> pos
5
```

(2.91)

Now, we can write the **RosterToBit** procedure.

```
1 RosterToBit := proc(roster, universe)
2   local B, i, e, pos;
3   B := [seq(0, i=1..nops(universe))];
4   for e in roster do
5     if member(e[1], universe, 'pos') then
6       B[pos] := e[2];
7     else
8       error "Roster contained a member not in the universe.";
9     end if;
10  end do;
11  return B;
12 end proc;
```

```
> RosterToBit(fuzzyR, fuzzyU)
[0.3, 0.7, 0, 0, 0.1, 0, 0]
```

(2.92)

Observe that we surrounded the modification of **B** in an if statement to ensure that the roster does not contain any members not in the given universe.

### Converting between MultiSet and Roster Representations

Recall that even though a generalized **MultiSet** is displayed as if it were the same as our roster representation, it is actually not a set. However, the display suggests, and it is true, that converting a **MultiSet** into a list of pairs of objects and their multiplicities is easy. Simply apply the function **Entries**.

> *Entries*(fuzzyM)  
 {[“a”, 0.3], [“b”, 0.7], [“e”, 0.1]} (2.93)

In the other direction, we simply apply **MultiSet** to the roster representation.

> *MultiSet[generalized]*(fuzzyR)  
 {[“a”, 0.3], [“b”, 0.7], [“e”, 0.1]} (2.94)

While the usual syntax for the membership specification of a **MultiSet** is a sequence of such pairs, not a set of them, the command will automatically convert a set or list of object–degree pairs. Also note that **MultiSet** will convert a set of objects into a multiset in which each object has degree of membership equal to 1, and will convert a list into a multiset in which each object’s degree of membership is its multiplicity in the list.

## 2.3 Functions

In this section, we will see three different ways to represent functions in Maple and explore a variety of the concepts described in the text relative to these different representations.

### *Procedures*

In this manual, we have already seen several examples of procedures. In some ways, a procedure, or more broadly any computer program, is the ultimate generalization of a mathematical function. As an example, consider the **GetVars** procedure. This procedure assigns to each valid input (a Maple expression) a unique output (a list of the names appearing in the expression). Setting  $A$  equal to the set consisting of all possible Maple expressions and  $B$  equal to all possible lists of valid names, **GetVars** satisfies the definition of being a function from  $A$  to  $B$ .

We discussed procedures in some depth in the introductory chapter. Here, we will discuss the concepts of domain and codomain as they relate to programs via the computer programming concept of type.

In Example 5 of Section 2.3, the text gives examples from Java and C++ showing how domain and codomain are specified in those programming languages. The procedure below illustrates how this is done in Maple.

```

1 Floor1 := proc (x : float) : integer;
2   return floor(x);
3 end proc;
```

The body of our **Floor1** procedure is merely a call to Maple’s internal **floor** command. The focus of the example is to illustrate how you can specify the domain (i.e., the type of a parameter) of a procedure and the codomain (i.e., the return type). Note that in Maple, unlike some languages such as Java and C++, such declarations are entirely optional.

Declaring the return type of a procedure is done by following the right parenthesis that ends the list of parameters with two colons and a valid Maple type and then a semicolon. Typically, this has no effect on the actual operation of the procedure and is more informational. It is possible to have Maple enforce the return type by executing the command **kernelopts(assertlevel) := 2;**

More information can be found on the **proc** help page. In this manual, we will typically not declare return types for procedures.

To declare the type of parameters, you follow the name of the parameter by two colons and a valid Maple type. When the procedure is applied, before any of the code in the body of the procedure is executed, Maple checks the input against the declared type. If the input does not match, then an error is raised.

```
> Floor1("hello")
```

Error, invalid input: Floor1 expects its 1st argument, x, to be of type float, but received hello

This is useful because it helps to ensure that the procedure is never applied to invalid input, which may have undesirable consequences. For example, consider the procedure below, which prints and then decreases its argument by 1 until it reaches 0.

```
1 loopy := proc (n : posint)
2     local m;
3     m := n;
4     while m <> 0 do
5         print(m);
6         m := m - 1;
7     end do;
8 end proc;
```

Applying this procedure with a positive integer as its argument has the desired result.

```
> loopy(3)
3
2
1
0
```

(2.95)

If you call the function with a value that is not of type **posint**, it will raise an error.

```
> loopy(-5)
```

Error, invalid input: loopy expects its 1st argument, n, to be of type posint, but received -5

Without the parameter declared as a positive integer, applying **loopy** to  $-5$  would have resulted in an infinite loop.

Some common types are: float, integer, posint, nonnegint, set, and list. A complete list can be found on the help page for type. In addition, Maple provides a method for easily creating new types via the structured type syntax. For example, our **Floor1** procedure has a slight problem, as the following illustrates.

```
> Floor1(5.)
5
```

(2.96)

> *Floor1*(5)

Error, invalid input: Floor1 expects its 1st argument, x, to be of type float, but received 5

Maple distinguishes integers like **5** from floats like **5.**. We can make our procedure accept either floats or integers as follows.

```
1 | Floor2 := proc (x : {float, integer})
2 |     return floor(x);
3 | end proc;
```

> *Floor2*(5.), *Floor2*(5)  
5, 5 (2.97)

Enclosing two or more types in braces indicates that any of the types are acceptable.

It is also useful to be able to specify that a procedure should accept a set or list. We rewrite the **Floor** procedure once again so that it accepts a list and applies Maple's **floor** function to each of the members of the list.

```
1 | Floor3 := proc (L : list)
2 |     return map(floor, L);
3 | end proc;
```

> *Floor3*([12.5, 13.9, -2.5])  
[12, 13, -3] (2.98)

Note that we have used the **map** command to apply the **floor** function to each member of the input list **L**. The **map** command has a variety of syntax options, but this is one of the more common.

We can also be more specific and specify that the list must contain floats and integers. We do this by following the list keyword with parentheses and indicating the types that are allowed in the list.

```
1 | Floor4 := proc (L : list({float, integer}))
2 |     return map(floor, L);
3 | end proc;
```

> *Floor4*([47.298, 3, 13.7])  
[47, 3, 13] (2.99)

> *Floor4*([5.6, 3,  $\frac{22}{7}$ , 6.8])

Error, invalid input: Floor4 expects its 1st argument, L, to be of type list({float, integer}), but received [5.6, 3, 22/7, 6.8]

> *Floor4*(3.2)

Error, invalid input: Floor4 expects its 1st argument, L, to be of type list({float, integer}), but received 3.2

The second and third applications of **Floor4** failed because **22/7** is neither a float nor an integer and **3.2** is not a list. Maple provides the numeric type as a useful catch-all for numeric objects. We can make our procedure accept either single values or lists as follows.

```
1 Floor := proc (v : { numeric, list ( numeric ) } )
2     if type (v, numeric) then
3         return floor (v) ;
4     else
5         return map (floor, v) ;
6     end if ;
7 end proc;
```

> *Floor*  $\left([5.6, 3, \frac{22}{7}, 6.8]\right)$   
[5, 3, 3, 6] (2.100)

> *Floor*(3.2)  
3 (2.101)

## Functional Operators

Next, we will look at functional operators as a way to represent functions in Maple. This is the most natural representation for functions defined by a formula.

To represent the function defined by the formula  $f(x) = x^2$ , we enter the following command.

> *f* := x -> x^2;  
*f* :=  $x \mapsto x^2$  (2.102)

There are three basic components to defining a functional operator. First, the name, which in this case is **f**. The name is followed by the assignment operator. Second is the variable or variables. If you wish the function to accept multiple variables, they must be enclosed in parentheses. Following the variables, you enter an arrow composed of a hyphen and then a greater than symbol. And third is the formula that provides the result. (Note that we presented this first example in 1-D input mode; when using 2-D mode, the hyphen–greater-than pair is converted into an arrow as you type.)

You apply the functional operator just as you would expect.

> *f*(3)  
9 (2.103)



## Composition of Functions

We will now use functional operators to briefly explore composition and graphs of functions. You can combine functions algebraically in the natural way. Maple will return the formula for a functional operator when you apply it to an unassigned name.

$$\begin{aligned} > g := x \rightarrow x + 1 \\ &g := x \mapsto x + 1 \end{aligned} \tag{2.104}$$

$$\begin{aligned} > (f + g)(x) \\ &x^2 + x + 1 \end{aligned} \tag{2.105}$$

$$\begin{aligned} > \left( \frac{f}{g} \right)(t) \\ &\frac{t^2}{t + 1} \end{aligned} \tag{2.106}$$

The parentheses around **f+g** and **f/g** are necessary because the application of a procedure is of higher precedence than the arithmetic operators. Contrast the above with **f/g(t)**, which results in the name **f** divided by the formula for **g**.

$$\begin{aligned} > f/g(t); \\ &\frac{f}{t + 1} \end{aligned} \tag{2.107}$$

For composition, Maple provides the **@** operator.

$$\begin{aligned} > (g @ f)(x) \\ &x^2 + 1 \end{aligned} \tag{2.108}$$

Composition can also be applied to procedures, both those defined by you and any built into Maple. For example, the following defines the function that computes the square of the floor of a number.


$$\begin{aligned} > \text{squareFloor} := f @ \text{floor} \\ &\text{squareFloor} := f @ \text{floor} \end{aligned} \tag{2.109}$$

$$\begin{aligned} > \text{squareFloor}(3.2) \\ &9 \end{aligned} \tag{2.110}$$

## Plotting Graphs of Functions


You can use Maple to draw the graph of a function by using the **plot** command. The most basic syntax requires only two arguments: the function to be graphed in terms of an independent variable and the variable. The first argument can be given as an expression as follows.

$$> \text{plot}(x^2, x)$$




The first argument can also be given as a functional operator or even a procedure as the following two examples illustrate. The essential requirement is that the first argument must evaluate to a numeric value whenever the independent variable is assigned a value. That is to say, the function must have appropriate domain and range.

*> plot(f(x), x)*




*> plot(floor(x), x)*



In the plot of **floor**, the vertical lines are artifacts of how Maple draws graphs. Basically, Maple is computing the value of the function at a large number of  $x$ -values between  $-10$  and  $10$ . It then “connects the dots.” The vertical lines appear when Maple connects the points from either side of the jumps. We can eliminate them with the **discont=true** option to indicate that the graph has discontinuities. We can also specify the range of  $x$ -values we want to display by entering the equation  **$x=\text{min}..\text{max}$**  as the second argument. In the next graph, we use those two options to display a graph of the square of the floor function.

```
> plot(squareFloor(x), x = -5..5, discont = true)
```



## Tables

For finite domains, a table can be used to represent a function. To define a table, use the **table** command. For example, suppose  $f$  is the function whose domain is the set of students in a class and that maps each student to their grade on an exam. Let  $f$  be defined by  $f(Ann) = 83$ ,  $f(Bob) = 79$ ,  $f(Carlos) = 91$ , and  $f(Delia) = 72$ . We model  $f$  as a table named **exams** by applying the **table** command to the list whose entries are equations of the form **a=b** to represent  $f(a) = b$ .

```
> exams := table(["Ann" = 83, "Bob" = 79, "Carlos" = 91, "Delia" = 72])
exams := table(["Ann" = 83, "Bob" = 79, "Carlos" = 91, "Delia" = 72]) (2.111)
```

Once the table is defined, you can obtain the value  $f(Carlos)$  with the bracket selection operation.

```
> exams["Carlos"]
91 (2.112)
```

You can also use selection together with assignment to modify values or to add entries to the table.

```
> exams["Ann"] := 84
exams["Ann"] := 84 (2.113)
```

```
> exams["Ernie"] := 86
exams["Ernie"] := 86 (2.114)
```

To see the table definition, you apply the **op** command. You can obtain the list of equations by applying **op** twice.

```
> exams
exams (2.115)
```

```
> op(exams)
table(["Ann" = 84, "Ernie" = 86, "Bob" = 79, "Carlos" = 91, "Delia" = 72]) (2.116)
```

```
> op(op(exams))
["Ann" = 84, "Ernie" = 86, "Bob" = 79, "Carlos" = 91, "Delia" = 72] (2.117)
```

## Domain and Range

Since tables are finite, we can write procedures to check various properties. First, we find the domain (that is, the domain of definition) and range of a function defined as a table. For these procedures, we use the commands **indices** and **entries**. In the **exams** table, the students' names are the *indices* or *keys* of the table and the scores (84, 79, etc.) are the *entries* or *values* of the table.

You would probably expect **indices(exams)** to return the list or set of students' names and **entries(exams)** to return the list of scores. Observe what actually is returned.

```
> indices(exams)
["Ann"], ["Ernie"], ["Bob"], ["Carlos"], ["Delia"] (2.118)
```

```
> entries(exams)
[84], [86], [79], [91], [72] (2.119)
```

The **indices** command returns a sequence of lists where each list contains an index of the table. The reason for this is to allow very complicated indices which may even be sequences of values. For example, we can define the following table.

```
> M := table():
> M[1,1] := 1:
> M[1,2] := 0:
> M[2,1] := 0:
> M[2,2] := 1:
> op(M)
table([(1,1) = 1, (1,2) = 0, (2,1) = 0, (2,2) = 1])
```

(2.120)

```
> indices(M)
[1,1], [1,2], [2,1], [2,2]
```

(2.121)

If the indices command had returned a sequence of the indices, it would have appeared that the indices 1 and 2 were repeated several times.

```
> (2,1), (1,1), (2,2), (1,2)
2, 1, 1, 1, 2, 2, 1, 2
```

(2.122)

(Note that we defined **M** to be the empty table and then added entries to it. This is a very common way to define a table. Readers with some experience with matrices will note that **M** is a representation of the identity matrix of dimension 2.)

In cases where you are sure that there is no need for **indices** to return the indices in lists, you can use the **nolist** symbol. The same is true for the **entries** command.

```
> indices(exams,nolist)
"Ann", "Ernie", "Bob", "Carlos", "Delia"
```

(2.123)

```
> entries(exams,nolist)
84, 86, 79, 91, 72
```

(2.124)

Note that while Maple chooses the order to report the indices and entries and the user has no control over that order, the order is consistent between the two results; that is, the entry listed first is the entry corresponding to the index listed first, the second entry corresponds to the second index, and so on.

The discussion above allows us to easily write procedures to compute the domain and range of a function represented by a table.

```
1 FindDomain := proc(T : table)
2     return {indices(T,'nolist')};
3 end proc:
4 FindRange := proc(T : table)
```

```

5   return {entries(T,'nolist')};
6   end proc:

```

```

> FindDomain(exams)
    {"Ann", "Bob", "Carlos", "Delia", "Ernie"}

```

(2.125)

```

> FindRange(exams)
    {72, 79, 84, 86, 91}

```

(2.126)

### Injective and Surjective

We will create a few more examples and then write procedures to check for injectivity and surjectivity. The examples below correspond to the functions  $f_1(x) = x^2$ ,  $f_2(x) = x^3$ , and  $f_3(x) = |x|$  on the domain  $D = \{-5, -4, 5, \dots\}$ .

Thus far, we have defined tables by manually creating all of the index–entry pairs. Here, we will use **seq** to form the equations that define the table. We simply provide the equation as the first argument.

```

> f1 := table([seq(x = x^2, x = -5..5)])
    f1 := table([-1 = 1, 0 = 0, -2 = 4, 1 = 1, -3 = 9, 2 = 4,
                -4 = 16, 3 = 9, 4 = 16, -5 = 25, 5 = 25])

```

(2.127)

```

> f2 := table([seq(x = x^3, x = -5..5)])
    f2 := table([-1 = -1, 0 = 0, -2 = -8, 1 = 1, -3 = -27,
                2 = 8, -4 = -64, 3 = 27, 4 = 64, -5 = -125, 5 = 125])

```

(2.128)

```

> f3 := table([seq(x = |x|, x = -5..5)])
    f3 := table([-1 = 1, 0 = 0, -2 = 2, 1 = 1, -3 = 3, 2 = 2,
                -4 = 4, 3 = 3, 4 = 4, -5 = 5, 5 = 5])

```

(2.129)

The domain and range functions defined earlier produce the expected results.

```

> FindDomain(f1)
    {-5, -4, -3, -2, -1, 0, 1, 2, 3, 4, 5}

```

(2.130)

```

> FindRange(f1)
    {0, 1, 4, 9, 16, 25}

```

(2.131)

We can check to see if a table is surjective for a specified codomain by comparing the codomain to the range.

```

1   IsOnto := proc(T :: table, codomain :: set)
2       return evalb(FindRange(T) = codomain);
3   end proc:

```

```

> IsOnto(f1, {0, 1, 2, 3, 4, 5})
    false

```

(2.132)

```

> IsOnto(f3, {0, 1, 2, 3, 4, 5})
    true

```

(2.133)

We can check for injectivity by making sure that no entry value is repeated. The easiest way to do this is to check that the number of values in the result of **FindRange** is the same as the number in the domain returned by **FindDomain**.

```
1 IsOnetoOne := proc (T : table)
2   return evalb(nops(FindDomain(T)) = nops(FindRange(T)));
3 end proc;
```


> *IsOnetoOne(f1)*  
*false* (2.134)

> *IsOnetoOne(f2)*  
*true* (2.135)

### Graphing a Function from a Table


Finally, we see how to graph a function defined by a table. An alternative form of the **plot** command accepts two arguments: the first argument is the list of  $x$ -values and the second argument is the list of  $y$ -values. For example,

> *plot([1, 2, 3], [2, 0, 1])*




Note that Maple connected the points (1, 2), (2, 0), and (3, 1) to draw the graph. We can use the **style=point** option to draw points instead of connecting the dots. In addition, the options **symbol=solidcircle** and **symbolsize=15** will cause the points to be drawn as solid circles 15 points in diameter. Finally, **view=[0..4,-1..3]** will make the bounds of the graph 0 to 4 on the  $x$  axis and  $-1$  to 3 on the  $y$ .

> *plot([1, 2, 3], [2, 0, 1], style = point, symbol = solidcircle, symbolsize = 15,  
view = [0..4, -1..3])*




Graphing a function defined by a table can be done in the same way, using the **indices** and **entries** commands to create lists of the  $x$  and  $y$  coordinates of the desired points. Remember that Maple orders the indices and entries so that they are consistent, that is, the  $x$  and  $y$  values will correspond.

```
> plot([indices(f1, nolist)], [entries(f1, nolist)], style = point,
      symbol = solidcircle, symbolsize = 15, view = [-5 .. 5, 0 .. 25])
```






```
> plot([indices(f2,nolist)],[entries(f2,nolist)],style = point,  
       symbol = solidcircle,symbolsize = 15,view = [-5..5,-125..125])
```



```
> plot([indices(f3,nolist)],[entries(f3,nolist)],style = point,  
       symbol = solidcircle,symbolsize = 15,view = [-5..5,0..5])
```



## Some Important Functions

We have already seen that Maple has a built-in floor function, **floor**. It also includes **ceil** for computing the ceiling of a real number.

```
> floor(2.7)
2
```

(2.136)

```
> ceil(2.7)
3
```

(2.137)

There are some additional functions related to those two. The **round** command rounds a number to the nearest integer. The **trunc** command truncates the number, removing any fractional part, producing the next nearest integer toward 0. And the **frac** command returns the fractional part of the number.

```
> round(2.7)
3
```

(2.138)

```
> trunc(2.7)
2
```

(2.139)

```
> trunc(-2.7)
-2
```

(2.140)

```
> frac(2.7)
0.7
```

(2.141)

The text also discusses the factorial function. In Maple, you compute the factorial of a number by entering the number followed by the exclamation point. You can also use the **factorial** command.

```
> 6!
720
```

(2.142)

```
> factorial(6)
720
```

(2.143)

## 2.4 Sequences and Summations

In this section, we will see how Maple can be used to create and manipulate sequences, and in particular, we will see a way to use Maple to generate the terms of a recurrence sequence. We will also look at summations and see how Maple's symbolic computation abilities can be used to explore both finite and infinite series.

Sequences are fundamental to Maple. In Maple, an expression sequence is any ordered collection of valid expressions separated by commas. For example,

```
> aSequence := 1, "a", x, Pi, 3 x^2 + 5, {"a", "b", "c"}
aSequence := 1, "a", x, pi, 3 x^2 + 5, {"a", "b", "c"}
```

(2.144)

is an expression sequence (or just sequence). Note that both sets and lists are formed by wrapping an expression sequence in the appropriate symbols, and procedures are called on particular values by passing the procedure a sequence of arguments in parentheses.

Elements of a sequence can be accessed in the same way as lists and sets, with the selection operation, as follows.

```
> aSequence3
x
```

(2.145)

The **nops** command cannot be used on a sequence in order to determine its length. To find the number of elements in a sequence, you must first convert the sequence to a list and then apply **nops**. Likewise, **op** does not work correctly for sequences.

```
> nops(aSequence)
```

Error, invalid input: nops expects 1 argument, but received 6

```
> nops([aSequence])
6
```

(2.146)

```
> op(aSequence)
```

Error, invalid input: op expects 1 or 2 arguments, but received 6

Many commands commonly used with lists and sets will produce errors or incorrect results when applied to a sequence.

The “empty sequence” is represented by the name **NULL**. In procedures, we will often initialize a variable to **NULL** in order to build up a sequence of values. You may also have a procedure return **NULL** in order to cause the procedure to exit without displaying output.

There are three main tools for creating a sequence in Maple: the comma operator, the **seq** command, and the **\$** operator.

### ***Building Sequences: The Comma Operator***

The comma operator is used to join two expressions or expression sequences into a sequence. The comma operator is used when forming a sequence by listing the elements, as in the following.

```
> sequence2 := 1, 2, 3, 4
sequence2 := 1, 2, 3, 4
```

(2.147)

It is also used to combine two existing sequences or a sequence and a single expression. Note that the original sequences are not modified unless you reassign the name to the result.

```
> aSequence, sequence2
1, “a”, x,  $\pi$ ,  $3x^2 + 5$ , {“a”, “b”, “c”}, 1, 2, 3, 4
```

(2.148)

```
> sequence2 := sequence2, 15
sequence2 := 1, 2, 3, 4, 15
```

(2.149)

This application of the comma operator is often used in procedures to build a sequence one term at a time. For example, the following procedure builds the sequence consisting of the first  $n + 1$  terms of the geometric progression  $a, ar, ar^2, ar^3, \dots, ar^n$ .

```

1  GeometricSeq := proc (a, r, n)
2      local S, i;
3      S := NULL;
4      for i from 0 to n do
5          S := S, a*r^i;
6      end do;
7      return S;
8  end proc;
```

> *GeometricSeq*(3, 4, 10)  
 3, 12, 48, 192, 768, 3072, 12 288, 49 152, 196 608,  
 786 432, 3 145 728 (2.150)

In that procedure, the output sequence **S** is initialized to the **NULL** value. At each step in the for loop, the next term in the sequence is added to the existing sequence **S** with the comma operator.

### ***Building Sequences: The seq Command***

We have already seen several examples of the **seq** command. We briefly summarize some of the ways it can be called.

The most common way to call **seq** is demonstrated in the following example, which recreates the geometric sequence produced above.

> *seq*( $3 \cdot 4^i, i = 0..10$ )  
 3, 12, 48, 192, 768, 3072, 12 288, 49 152, 196 608,  
 786 432, 3 145 728 (2.151)

The first argument is an expression which may involve an index variable, in this case **i**. The second argument is of the form **i=m..n**. This indicates that the index variable should range from **m** to **n**.

A third argument can be added to control the step, that is, the amount by which the index variable is incremented. For example, the command below will produce every other term of the geometric sequence from above.

> *seq*( $3 \cdot 4^i, i = 0..10, 2$ )  
 3, 48, 768, 12 288, 196 608, 3 145 728 (2.152)

The bounds of the range for the index and the step do not necessarily need to be integers. For example,

> *seq*( $i, i = 2.3..5.6, 0.25$ )  
 2.3, 2.55, 2.80, 3.05, 3.30, 3.55, 3.80, 4.05, 4.30, 4.55, 4.80,  
 5.05, 5.30, 5.55 (2.153)

There is also an abbreviated form allowing you to omit the first argument and the index variable. For example, to obtain the first ten positive even integers, you can issue the following command.

```
> seq(2..20,2)
2, 4, 6, 8, 10, 12, 14, 16, 18, 20
```

(2.154)

The other main use of **seq** is to apply the expression to each element of a set or list. This is illustrated in the command below which finds the squares of the first six prime numbers.

```
> seq(i^2, i = {2, 3, 5, 7, 11, 13})
4, 9, 25, 49, 121, 169
```

(2.155)

Note that for sets, the order of the elements in the sequence is determined by the order that Maple imposes on the set. For example, if we rearrange the primes in the example above, the output will be the same.

```
> seq(i^2, i = {2, 3, 5, 7, 11, 13})
4, 9, 25, 49, 121, 169
```

(2.156)

To impose a particular order, use a list instead.

```
> seq(i^2, i = [2, 5, 7, 13, 11, 3])
4, 25, 49, 169, 121, 9
```

(2.157)

As an alternative to **i=**, you may use the word **in** in place of the equals sign.

```
> seq(i^2, i in [2, 5, 7, 13, 11, 3])
4, 25, 49, 169, 121, 9
```

(2.158)

While **seq** is most commonly used in conjunction with lists and sets, any expression can be used in place of the list. For example, the following computes the sequence consisting of the squares of the terms in the given algebraic expression.

```
> seq(i^2, i = 3x^5 + 2x^4 - x^3 + 7x^2 - 8x + 9)
9x^10, 4x^8, x^6, 49x^4, 64x^2, 81
```

(2.159)

In fact, **seq** can be used to iterate over any expression to which **op** can be applied.

### ***Building Sequences: The \$ Operator***

The **\$** operator is an alternative to the **seq** command, though it is somewhat limited. The **\$** operator is a binary operator, like **+** or **\***. Its left operand is the expression in terms of an index variable and the right operand is the equation that specifies the range for the variable. We can produce the geometric sequence from above as follows.

```
> 3 * 4^k $ k = 0..10
3, 12, 48, 192, 768, 3072, 12288, 49152, 196608,
786432, 3145728
```

(2.160)

Note that this is the same as **seq(3\*4^k,k=0..10)**.

It is sometimes necessary to enclose the expression on the left and the variable on the right in single right quotes to prevent early evaluation. This is the case when the index has previously been assigned a value.

```
> anIndex := 20;
   anIndex := 20
```

(2.161)

```
> anIndex · 2 + 3 $ anIndex = 1 .. 5
```

Error, invalid input: '\$' expects its 2nd argument, range, to be of type {numeric, algebraic, name = literal .. literal, name = algebraic .. algebraic}, but received 20 = 1 .. 5

The error occurs because Maple evaluates the expressions on both sides of the \$ operator before applying the operator to them. Since **anIndex** stored a value, the right-hand side became **20=1..5**, causing an error. We fix this by enclosing the name of the index variable on the right in single quotes so that it is not evaluated. We also must enclose the entire expression on the left in single quotes so that it is not evaluated to 43.

```
> 'anIndex · 2 + 3' $ 'anIndex' = 1 .. 5
   5, 7, 9, 11, 13
```

(2.162)

The single quotes were not necessary for previous examples because **k** had not been assigned a value. Note that this issue is not a concern for **seq**.

```
> seq(anIndex · 2 + 3, anIndex = 1 .. 5)
   5, 7, 9, 11, 13
```

(2.163)

The \$ operator is particularly useful in the following two situations. First, like **seq**, the left operand and the index variable can be omitted to produce the specified range.

```
> $ 5 .. 11
   5, 6, 7, 8, 9, 10, 11
```

(2.164)

Second, if the right operand is an integer rather than a range, \$ will produce that many copies of the right operand. For example, to create a sequence of 11 copies of the string “a”, we give “a” as the left-hand operand and 11 as the right operand to \$.

```
> “a” $ 11
   “a”, “a”, “a”, “a”, “a”, “a”, “a”, “a”, “a”, “a”, “a”
```

(2.165)

Doing this with the **seq** command requires the following.

```
> seq(“a”, i = 1 .. 11)
   “a”, “a”, “a”, “a”, “a”, “a”, “a”, “a”, “a”, “a”, “a”
```

(2.166)

## Recurrence Relations

Next, we will see how we can use Maple to explore sequences that arise from recurrence relations. We will go into much more depth, especially in regards to Maple’s functions related to solving

recurrence relations, in Chapter 8. Here, we will only explore how we can have Maple compute terms of sequences defined by recurrence relations.

As an example, consider the Fibonacci sequence, which has recurrence relation  $f_n = f_{n-1} + f_{n-2}$  and initial conditions  $a_1 = 0$  and  $a_2 = 1$ . (Note that the text uses 0 as the first index for a sequence, but Maple uses 1 as the first index for sequences and lists, and we will follow the Maple convention.) To produce this sequence in Maple, we can use a functional operator to represent the recurrence relation as follows.

$$\begin{aligned} &> \text{Fib} := n \rightarrow \text{Fib}(n-1) + \text{Fib}(n-2) \\ &\quad \text{Fib} := n \mapsto \text{Fib}(n-1) + \text{Fib}(n-2) \end{aligned} \tag{2.167}$$

Next, we set the initial values as follows.

$$\begin{aligned} &> \text{Fib}(1) := 0 \\ &\quad \text{Fib}(1) := 0 \end{aligned} \tag{2.168}$$

$$\begin{aligned} &> \text{Fib}(2) := 1 \\ &\quad \text{Fib}(2) := 1 \end{aligned} \tag{2.169}$$

Now, Maple will compute values of the sequence.

$$\begin{aligned} &> \text{Fib}(7) \\ &\quad 8 \end{aligned} \tag{2.170}$$

To display the sequence, use the **seq** command.

$$\begin{aligned} &> \text{seq}(\text{fib}(n), n = 1 .. 20) \\ &\quad 0, 1, 1, 2, 3, 5, 8, 13, 21, 34, 55, 89, 144, 233, 377, 610, 987, \\ &\quad 1597, 2584, 4181 \end{aligned} \tag{2.171}$$

While the above approach for calculating recurrence relations is convenient and intuitive, it does not make available all of the facilities for improving efficiency that are available using the **proc** command. We can get Maple to calculate these values more efficiently by using the **remember** option. This option causes Maple to “remember” any values for the procedure that it has already computed by storing them in a table.

```

1  Fib2 := proc (n : posint)
2      option remember;
3      if n <= 2 then
4          return 1;
5      else
6          return Fib2(n-1) + Fib2(n-2);
7      end if;
8  end proc;
```

This procedure encompasses both the initial conditions (when  $n \leq 2$ ) and the recurrence formula. The **remember** option causes Maple to store the results of the procedure when it is called so that if it is called again with the same input, the result can be looked up in the “remember table” rather than recomputed.

To illustrate the difference in performance, let us see how long it takes to compute the 10 000th Fibonacci number.

$$\begin{array}{l} > \textit{st} := \textit{time}() : \quad \textit{Fib2}(10000) : \quad \textit{time}() - \textit{st} \\ \quad 0.030 \end{array} \qquad (2.172)$$

The **time** command returns the total CPU time used in the current Maple session. Thus, the above works by setting **st** (for start time) equal to the amount of CPU time used before executing the procedure being timed, then executing the procedure, and then computing the difference of the amount of CPU time used with the start time.

The output above shows the amount of time, in seconds, used to find the 10 000th Fibonacci number using **Fib2**. Note that if we repeat the computation,

$$\begin{array}{l} > \textit{st} := \textit{time}() : \quad \textit{Fib2}(10000) : \quad \textit{time}() - \textit{st} \\ \quad 0.007 \end{array} \qquad (2.173)$$

the total time taken drops considerably. This is because Maple does not need to compute the value again. You can cause Maple to reset the remember table with the **forget** command.

In comparison, consider the **Fib** functional operator applied to 30.

$$\begin{array}{l} > \textit{st} := \textit{time}() : \quad \textit{Fib}(30) : \quad \textit{time}() - \textit{st} \\ \quad 0.516 \end{array} \qquad (2.174)$$

Note that the purely recursive implementation **Fib** cannot be used to compute the 10 000th Fibonacci number. In fact, to compute the 10 000th Fibonacci number, **Fib** would need to be invoked approximately

$$\begin{array}{l} > \textit{Fib2}(9999) \\ \quad 2.08 \times 10^{2089} \end{array} \qquad (2.175)$$

times in order to handle all the recursive sub-calls that are made. (The reader is encouraged to prove this fact.)

Even at a billion calls per second, this would require

$$\begin{array}{l} > \frac{(2.175)}{1000000000.} \\ \quad 2.079360824 \times 10^{2080} \end{array} \qquad (2.176)$$



seconds, or

$$\begin{aligned} &> \frac{(2.176)}{60 \cdot 60 \cdot 24 \cdot 365} \\ &\quad 6.593609919 \times 10^{2072} \end{aligned} \tag{2.177}$$

years to complete.

## Summations

Finally, we will see how Maple can be used to compute with summations, both numerically for finite sums and symbolically for infinite sums.

To add a finite sequence of values, we use the **add** command. This command is very similar to the **seq** command, though with somewhat fewer options. It requires two arguments. The first argument must be an expression in terms of an index variable such as **i**. The second argument can be either an equation of the form **i=m..n**, indicating the range of values for the index variable, or it can be of the form **i=x** or **i in x** where **x** is a list, set, or other such object. The forms **i=x** and **i in x** are equivalent.

For example, to compute the sum of the squares of the first ten positive integers,  $\sum_{i=1}^{10} i^2$ , we enter the following.

$$\begin{aligned} &> \text{add}(i^2, i = 1..10) \\ &\quad 385 \end{aligned} \tag{2.178}$$

To compute the sum of the members of a list, you can enter:

$$\begin{aligned} &> \text{add}(i, i \text{ in } [1, 2, 4, 6, 9, 11, 14]) \\ &\quad 47 \end{aligned} \tag{2.179}$$

This can be shortened by providing the list as the sole argument.

$$\begin{aligned} &> \text{add}([1, 2, 4, 6, 9, 11, 14]) \\ &\quad 47 \end{aligned} \tag{2.180}$$

The **mul** command is used with the same syntax as **add** to compute products of sequences.

The **sum** command is used for symbolic summation. The first argument to the **sum** command is the same as for **add**, an expression in terms of an index variable. The second argument is of the form **i=m..n**. The main difference is that for **add**, **m** and **n** must be numbers, while for **sum**, they can be unassigned names or even **infinity**.

As an example, we have Maple compute the sum of the squares of the first  $n$  positive integers,  $\sum_{k=1}^n k^2$ .

$$\begin{aligned} &> \text{sum}(k^2, k = 1..n) \\ &\quad \frac{(n+1)^3}{3} - \frac{(n+1)^2}{2} + \frac{n}{6} + \frac{1}{6} \end{aligned} \tag{2.181}$$

We can also compute the sum of the terms with even index up to  $2n$  in a geometric series, that is,

$$\sum_{k=0}^n ar^{2k}.$$

$$> \text{sum}(a \cdot r^{2 \cdot k}, k = 0 .. n)$$

$$\frac{a(r^2)^{n+1}}{r^2 - 1} - \frac{a}{r^2 - 1} \quad (2.182)$$

Finally,  $\sum_{k=1}^{\infty} kx^{k-1}$ , with  $|x| < 1$ , is computed by

$$> \text{sum}(k \cdot x^{k-1}, k = 1 .. \text{infinity}) \text{ assuming } |x| < 1$$

$$\frac{1}{(x - 1)^2} \quad (2.183)$$

Note the keyword **assuming** followed by a condition expressing the radius of convergence. You can confirm that these results match the formulas given in Table 2 of Section 2.4.

## 2.5 Cardinality of Sets

In this section, we explore the cardinality of the positive rational numbers. In Example 4 of Section 2.5 of the text, it is shown that the positive rationals are countable by describing how to list them all. Here, we will use Maple to implement this listing algorithm. We will also consider the following two questions. First, given a positive rational number, what is its position in the list? Second, given a positive integer, what fraction is located at that position within the list?

### *Enumerating the Positive Rationals*

We begin by reviewing the description in Example 4. The first element of the list is the rational number  $\frac{1}{1}$ . Second, we list the positive rationals  $\frac{p}{q}$  such that  $p + q = 3$ . Next come the rationals with  $p + q = 4$ , excluding  $\frac{2}{2}$  which is already in the list, being equivalent to  $\frac{1}{1}$ . This continues for each  $n$ : we list the fractions  $\frac{p}{q}$  such that  $p + q = n$ , excluding those equivalent to fractions already in the list.

In our procedure, we refer to  $n$  as the stage, so that in stage 5, for example, we are listing the fractions  $\frac{p}{q}$  such that  $p + q = 5$ . The stage  $n$  will range from 2 up to some maximum value. This maximum value of  $n$  will be the parameter to the procedure. We implement this as a for loop with index variable  $n$ .

Within each stage, that is, within the for loop, we need to generate the rational numbers  $\frac{p}{q}$  and add them to the list, provided they are not already in it. We can rewrite  $p + q = n$  as  $p = n - q$ . By allowing  $q$  to range from 1 to  $n - 1$  and calculating  $p$ , we will produce all the potential rationals in stage  $n$ . The **in** operator, discussed in Section 2.1 in relation to sets, applies to lists as well. Therefore, for each  $q$  from 1 to  $n - 1$ , we will form the fraction  $\frac{p}{q}$  (with  $p = n - q$ ), use **in** to test whether this is already in our list of positive rationals, and, if not, add it to the list.

Here is the complete procedure.

```

1 ListRationals := proc (max : posint)
2   local L, n, p, q;
3   L := [];
4   for n from 2 to max do
5     for q from 1 to n-1 do
6       p := n-q;
7       if not (p/q in L) then
8         L := [op(L), p/q];
9       end if;
10    end do;
11  end do;
12  return L;
13 end proc;

```

Applying this procedure to 6, we obtain the list through stage 6.

> *ListRationals* (6)

$\left[1, 2, \frac{1}{2}, 3, \frac{1}{3}, 4, \frac{3}{2}, \frac{2}{3}, \frac{1}{4}, 5, \frac{1}{5}\right]$

(2.184)

### *Finding the Position Given a Positive Rational*

Suppose we want to determine the position of a particular fraction within the list. Take for example  $29/35$ . Since  $29 + 35 = 64$ , we know that this fraction would first appear in stage 64. Thus, we compute the list up to stage 64.

> *RatsTo64* := *ListRationals*(64):

We suppress the output because this is a long list:

> *nops* (*RatsTo64*)

1259

(2.185)

Now, we work backwards from the end of the list until we find the desired fraction. A simple loop will help with this. Including a **by** clause in a for loop allows us to specify how much the index variable is changed each time, so **by -1** causes the for loop to step backwards by 1 for each iteration. Once we find the location of the desired fraction, we display the location and **break** the loop.

> **for** *i* **from** 1259 **by** -1 **to** 1 **do**

**if** *RatsTo64*[*i*] =  $\frac{29}{35}$  **then**

*print*(*i*);

**break**;

**end if**;

**end do**;

1245

(2.186)

We can make this process into a procedure. Given a fraction, the **numer** and **denom** commands will extract the numerator and denominator, respectively.

```
> numer( $\frac{29}{35}$ )
29
```

(2.187)

```
> denom( $\frac{29}{35}$ )
35
```

(2.188)

Our procedure can accept a rational number (type **rational**) as input with an additional check to make sure the input is positive. We sum the results of **numer** and **denom** to determine the stage. Within the for loop, we use a **return** statement instead of **print** and **break**.

```
1 LocateRational := proc (r : rational)
2   local stage, L, i;
3   if r <= 0 then
4     error "Input value must be positive .";
5   end if;
6   stage := numer(r) + denom(r);
7   L := ListRationals(stage);
8   for i from nops(L) to 1 by -1 do
9     if L[i] = r then
10      return i;
11    end if;
12  end do;
13 end proc;
```

```
> LocateRational( $\frac{75}{197}$ )
22 566
```

(2.189)

### ***Finding the Rational in a Given Position***

On the other hand, suppose we want to know which fraction is at a particular position. For instance, say we want to know which is the 100th fraction listed. If we knew which stage of the process would yield a list of at least 100 rational numbers, we could just generate the list up to that stage. We can guess and check until we found a stage that produced a long enough list.

### **Putting a Lower Bound on the Number of Stages**

We can guide our guesses a bit, however. Remember that at stage 2, the process generates 1 fraction. At stage 3, it generates 2 fractions. At stage 4, it generates 3 fractions, although one of them is discarded because it is a repeat. At stage  $k$ , the process generates  $k - 1$  rational numbers, some of which may be discarded as repeats. Therefore, we know that, after stage  $n$  is complete, the number of rational numbers in our list contains *at most*  $\sum_{k=2}^n k - 1$  rational numbers. We can use the **sum** command discussed in the previous section to find a formula for this summation.

$$\begin{aligned}
 &> \text{sum}(k-1, k=2..n) \\
 &\quad \frac{(n+1)^2}{2} - \frac{3n}{2} - \frac{1}{2}
 \end{aligned}
 \tag{2.190}$$

Applying **factor** will give us a more convenient formula.

$$\begin{aligned}
 &> \text{factor}((2.190)) \\
 &\quad \frac{n(n-1)}{2}
 \end{aligned}
 \tag{2.191}$$

In other words, the number of rational numbers in the list produced by **ListRationals** at the conclusion of stage  $n$  is at most  $\frac{n(n-1)}{2}$ . Define  $F(n)$  to be the number of positive rational numbers produced by the **ListRationals** algorithm at the conclusion of stage  $n$ . Alternatively,  $F(n)$  is the number of distinct positive rational numbers  $\frac{p}{q}$  such that  $p+q \leq n$ . Thus, we have determined that

$$F(n) \leq \frac{n(n-1)}{2}.$$

We now return to the question of how many stages we need to compute in order to find the 100th rational number. We can restate this as follows: find  $n$  such that  $F(n) \geq 100$ . Combining our inequalities, we have that  $\frac{n(n-1)}{2} \geq 100$ . Maple's **solve** command will solve the equation for us. (Note that since an approximation is sufficient, we will enter the 100 as **100.** so that Maple will solve using floating-point arithmetic.)

$$\begin{aligned}
 &> \text{solve}\left(\frac{n(n-1)}{2} = 100.\right) \\
 &\quad 14.65097170, -13.65097170
 \end{aligned}
 \tag{2.192}$$

This indicates that a stage of 14 *is not enough*, but it gives us a place to start guessing.

$$\begin{aligned}
 &> \text{RatsTo15} := \text{ListRationals}(15): \\
 &\quad \text{nops}(\text{RatsTo15}) \\
 &\quad 71
 \end{aligned}
 \tag{2.193}$$

$$\begin{aligned}
 &> \text{RatsTo17} := \text{ListRationals}(17): \\
 &\quad \text{nops}(\text{RatsTo17}) \\
 &\quad 95
 \end{aligned}
 \tag{2.194}$$

$$\begin{aligned}
 &> \text{RatsTo18} := \text{ListRationals}(18): \\
 &\quad \text{nops}(\text{RatsTo18}) \\
 &\quad 101
 \end{aligned}
 \tag{2.195}$$

> *RatsTo18*[100]

$$\frac{5}{13}$$

(2.196)

### How Tight is the Bound?

We just saw how the formula  $\frac{n(n-1)}{2}$  is an upper bound for  $F(n)$ , the number of positive rationals listed by the end of stage  $n$ . We conclude this section by exploring how good of a bound this is. In Section 2.3, we saw how to use the **plot** command to graph points. We use that technique to graph the upper bound up to  $n = 100$ .

To graph points, we need two lists for the  $x$  and  $y$  values to be graphed. The  $x$  values will be the values of  $n$ . We create the list by using the **seq** command.


> *xValues* := [*seq*( $n$ ,  $n = 2..100$ )] :

For the  $y$ -values, we use the **seq** command with the formula for the upper bound:  $\frac{n(n-1)}{2}$ .

> *boundValues* := [*seq*( $\frac{n(n-1)}{2}$ ,  $n = 2..100$ )] :

Now, we can plot the bound using the **plot** command and the options described in Section 2.3.

> *plot*(*xValues*, *boundValues*, *style* = *point*, *symbol* = *solidcircle*,  
    *symbolsize* = 15, *view* = [0..100, 0..5000])




To find the actual values of  $F(n)$ , we need the size of the list returned by **ListRationals** applied to  $n$ . In other words, we apply **nops** to the result of **ListRationals(n)**. Again, we use **seq** to form the list of these counts.

```
> actualValues := [seq(nops(ListRationals(n)), n = 2..100)] :
```

Again, we plot.

```
> plot(xValues, actualValues, style = point, symbol = solidcircle,
      symbolsize = 15, view = [0..100, 0..5000])
```




Since we used the same **view** for both graphs, you can see that the value of  $F(n)$  is much smaller than the upper bound. We can make the comparison easier by overlaying the graphs. We do this as follows. First, we assign the graphs to names. (We also change the color of the graph of the  $F(n)$  data to blue.)

```
> boundPlot := plot(xValues, boundValues, style = point,
  symbol = solidcircle, symbolsize = 15, view = [0..100, 0..5000],
  legend = "bound")

> actualPlot := plot(xValues, actualValues, style = point,
  symbol = solidcircle, symbolsize = 15, view = [0..100, 0..5000],
  color = blue, legend = "actual")
```

Finally, we have Maple draw the two plots together using the **display** command in the **plots** package.

```
> plots[display](boundPlot, actualPlot)
```



You will explore  $F(n)$  and the upper bound  $\frac{n(n-1)}{2}$  further in the exercises.

## 2.6 Matrices

Maple provides extensive support for calculating with matrices. We begin this section by describing a variety of ways to construct matrices in Maple. Then, we consider matrix arithmetic and operations on zero-one matrices.

### *Constructing Matrices*

Matrices are constructed using the **Matrix** command. This command can be used in several different forms and with a large variety of options, only some of which we will discuss here. For complete information, refer to the Maple help page.

### **Specifying Entries by Listing the Rows**

The simplest way to construct a matrix in Maple is by representing each row as a list.

```
> m1 := Matrix([[1, 2, 3], [4, 5, 6]])
```

$$m1 := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix} \quad (2.197)$$

In the above example, we passed one argument to the **Matrix** command: a list of lists where the inner lists are the rows of the matrix. This list of lists is referred to as the matrix initializer.

You can also explicitly set the size of the matrix by giving the number of rows and columns as the first two arguments to the **Matrix** command.



$$\begin{aligned}
 &> m2 := \text{Matrix}(2, 3, [[1, 2, 3], [4, 5, 6]]) \\
 m2 &:= \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix}
 \end{aligned} \tag{2.198}$$

Note that if the specified dimension is smaller than what is indicated by the initializer list, an error is generated.

$$> m3 := \text{Matrix}(2, 2, [[1, 2, 3], [4, 5, 6]])$$

Error, (in Matrix) initializer defines more columns (3) than column dimension parameter specifies (2)

However, if the specified dimension is larger, Maple will create the matrix of the desired size and fill the rest of the entries with 0s.

$$\begin{aligned}
 &> m4 := \text{Matrix}(3, 4, [[1, 2, 3], [4, 5, 6]]) \\
 m4 &:= \begin{bmatrix} 1 & 2 & 3 & 0 \\ 4 & 5 & 6 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}
 \end{aligned} \tag{2.199}$$

If only one dimension is given, Maple assumes that a square matrix of that size is desired.

$$\begin{aligned}
 &> m5 := \text{Matrix}(4, [[1, 2, 3], [4, 5, 6]]) \\
 m5 &:= \begin{bmatrix} 1 & 2 & 3 & 0 \\ 4 & 5 & 6 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}
 \end{aligned} \tag{2.200}$$

You can have Maple pad the matrix with a different value by using the optional **fill=value** argument. Below, we create a square matrix of dimension 3 whose entries are all 5.

$$\begin{aligned}
 &> m6 := \text{Matrix}(3, \text{fill} = 5) \\
 m6 &:= \begin{bmatrix} 5 & 5 & 5 \\ 5 & 5 & 5 \\ 5 & 5 & 5 \end{bmatrix}
 \end{aligned} \tag{2.201}$$

The initializer does not have to be a list of lists as in the previous examples. For instance, you can use another matrix, as in the example below where we expand the **m6** matrix.

$$\begin{aligned}
 &> m7 := \text{Matrix}(3, 4, m6, \text{fill} = 2) \\
 m7 &:= \begin{bmatrix} 5 & 5 & 5 & 2 \\ 5 & 5 & 5 & 2 \\ 5 & 5 & 5 & 2 \end{bmatrix}
 \end{aligned} \tag{2.202}$$

For very small matrices, you can use a shortcut notation, as illustrated below. The entries of the matrix are listed between a pair of angle brackets (< and >), with the entries in each row listed in order separated by commas and the rows separated by semicolons.

$$\begin{aligned}
 &> m8 := \langle 1, 2, 3, 4; 5, 6, 7, 8; 9, 10, 11, 12 \rangle \\
 m8 &:= \begin{bmatrix} 1 & 2 & 3 & 4 \\ 5 & 6 & 7 & 8 \\ 9 & 10 & 11 & 12 \end{bmatrix}
 \end{aligned} \tag{2.203}$$

If you prefer to enter the values as columns, you use vertical bars (|) in place of the semicolons.

$$\begin{aligned}
 &> m9 := \langle 1, 2, 3, 4 | 5, 6, 7, 8 | 9, 10, 11, 12 \rangle \\
 m8 &:= \begin{bmatrix} 1 & 5 & 9 \\ 2 & 6 & 10 \\ 3 & 7 & 11 \\ 4 & 8 & 12 \end{bmatrix}
 \end{aligned} \tag{2.204}$$

### Modifying Entries

Once a matrix has been created, its entries can be altered by assigning the new value to the specified location, with the square bracket selection notation used to indicate the desired location.

For example, to change the lower left entry of matrix  $m6$  to 6, you would enter the following command.

$$\begin{aligned}
 &> m6[3, 1] := 6 \\
 m6_{3,1} &:= 6
 \end{aligned} \tag{2.205}$$

Note that Maple reports the assignment. To see that it has happened, we have to explicitly command Maple to display the entire matrix.

$$\begin{aligned}
 &> m6 \\
 \begin{bmatrix} 5 & 5 & 5 \\ 5 & 5 & 5 \\ 6 & 5 & 5 \end{bmatrix}
 \end{aligned} \tag{2.206}$$

### Copying Matrices

Using a matrix as the initializer for **Matrix** is commonly used to make a copy of a matrix. Consider the following sequence of commands.

$$\begin{aligned}
 &> m7copy := m7 \\
 m7copy &:= \begin{bmatrix} 5 & 5 & 5 & 2 \\ 5 & 5 & 5 & 2 \\ 5 & 5 & 5 & 2 \end{bmatrix}
 \end{aligned} \tag{2.207}$$

$$\begin{aligned} > m7[1,2] := 11 \\ m7_{1,2} &:= 11 \end{aligned} \quad (2.208)$$

$$\begin{aligned} > m7 \\ \begin{bmatrix} 5 & 11 & 5 & 2 \\ 5 & 5 & 5 & 2 \\ 5 & 5 & 5 & 2 \end{bmatrix} \end{aligned} \quad (2.209)$$

$$\begin{aligned} > m7copy \\ \begin{bmatrix} 5 & 11 & 5 & 2 \\ 5 & 5 & 5 & 2 \\ 5 & 5 & 5 & 2 \end{bmatrix} \end{aligned} \quad (2.210)$$

Observe that the modification we made to the **m7** matrix was also made in the **m7copy** matrix. This is because the assignment **m7copy := m7** did not create a new copy of the matrix to store in **m7copy**. Instead, that assignment made both names refer to the same matrix. The assignment **m7[1,2] := 11** modified the row 1, column 2 entry of the unique matrix that both names refer to. In computer science, this is called a “reference type,” meaning that the name does not store the object, it stores a reference to the object. Assigning one name to another makes a copy of the reference, but both references refer to the same underlying object. Both matrices and tables are reference types in Maple.

To make a true copy of a matrix, you use the **Matrix** command with the matrix you want to copy as the initializer.

$$\begin{aligned} > m7realcopy &:= Matrix(m7) \\ m7realcopy &:= \begin{bmatrix} 5 & 11 & 5 & 2 \\ 5 & 5 & 5 & 2 \\ 5 & 5 & 5 & 2 \end{bmatrix} \end{aligned} \quad (2.211)$$

$$\begin{aligned} > m7[2,4] &:= 23 \\ m7_{2,4} &:= 23 \end{aligned} \quad (2.212)$$

$$\begin{aligned} > m7, m7copy, m7realcopy \\ \begin{bmatrix} 5 & 11 & 5 & 2 \\ 5 & 5 & 5 & 23 \\ 5 & 5 & 5 & 2 \end{bmatrix}, \begin{bmatrix} 5 & 11 & 5 & 2 \\ 5 & 5 & 5 & 23 \\ 5 & 5 & 5 & 2 \end{bmatrix}, \begin{bmatrix} 5 & 11 & 5 & 2 \\ 5 & 5 & 5 & 2 \\ 5 & 5 & 5 & 2 \end{bmatrix} \end{aligned} \quad (2.213)$$

While the modification of **m7** altered both **m7** and **m7copy**, **m7realcopy** was unchanged.

### Other Ways to Initialize Matrices

Another common way to create a matrix is by specifying the values with a single list rather than a list of lists. In this case, you must provide the dimension of the matrix, whereas it is optional if the initializer is a list of lists.

> *m10* := *Matrix*(2, 3, [1, 2, 3, 4, 5, 6])

$$m10 := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix} \quad (2.214)$$

You can also specify a matrix with a table. Note the use of parentheses around the table indices. In addition, in this case, the row and column dimensions are required.

> *m11table* := *table*([(1, 2) = 5, (1, 3) = 6, (2, 1) = -2])

$$m11table := \text{table}([(1, 2) = 5, (1, 3) = 6, (2, 1) = -2]) \quad (2.215)$$

> *m11* := *Matrix*(2, 3, *m11table*)

$$m11 := \begin{bmatrix} 0 & 5 & 6 \\ -2 & 0 & 0 \end{bmatrix} \quad (2.216)$$

As a shortcut, the data defining the table can be passed as a set to form the matrix.

> *Matrix*(2, 3, {(1, 2) = 5, (1, 3) = 6, (2, 1) = -2})

$$\begin{bmatrix} 0 & 5 & 6 \\ -2 & 0 & 0 \end{bmatrix} \quad (2.217)$$

Finally, you can initialize the matrix with a procedure. The procedure must accept two integers as arguments. The entries of the matrix are obtained by evaluating the procedure at the row and column number. Below we provide an example using a functional operator to make the entries in the matrix equal to the sum of the row and column numbers and then construct the  $4 \times 4$  matrix from it.

> *m12F* := (*i*, *j*) → *i* + *j*

$$m12F := (i, j) \mapsto i + j \quad (2.218)$$

> *m12* := *Matrix*(4, *m12F*)

$$m12 := \begin{bmatrix} 2 & 3 & 4 & 5 \\ 3 & 4 & 5 & 6 \\ 4 & 5 & 6 & 7 \\ 5 & 6 & 7 & 8 \end{bmatrix} \quad (2.219)$$

### ***Matrix Arithmetic***

The textbook defines addition and multiplication of matrices. Maple implements these operations on matrices in a fairly intuitive way. To add two matrices, you use the + operator, as you would expect.

> *m13* := *Matrix*([[1, 2, 3], [4, 5, 6]])

$$m13 := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix} \quad (2.220)$$

$$\begin{aligned}
 &> \text{m14} := \text{Matrix}([[-2, 3, -1], [1, 5, 2]]) \\
 \text{m14} &:= \begin{bmatrix} -2 & 3 & -1 \\ 1 & 5 & 2 \end{bmatrix}
 \end{aligned} \tag{2.221}$$

$$\begin{aligned}
 &> \text{m13} + \text{m14} \\
 &\begin{bmatrix} -1 & 5 & 2 \\ 5 & 10 & 8 \end{bmatrix}
 \end{aligned} \tag{2.222}$$

Maple's syntax for multiplying a matrix by a scalar is also intuitive.

$$\begin{aligned}
 &> 3 \text{m14} \\
 &\begin{bmatrix} -6 & 9 & -3 \\ 3 & 15 & 6 \end{bmatrix}
 \end{aligned} \tag{2.223}$$

This produces the matrix whose entries are three times the entries of **m12**.

Matrix multiplication is computed with the **.** (dot, typed as a period) operator instead of an asterisk. This is to emphasize that matrix multiplication is not commutative.

$$\begin{aligned}
 &> \text{m15} := \text{Matrix}([[3, 6, 11, 1], [-2, 5, 2, 0], [4, 8, 9, -3]]) \\
 \text{m15} &:= \begin{bmatrix} 3 & 6 & 11 & 1 \\ -2 & 5 & 2 & 0 \\ 4 & 8 & 9 & -3 \end{bmatrix}
 \end{aligned} \tag{2.224}$$

$$\begin{aligned}
 &> \text{m16} := \text{Matrix}([[2, 5], [1, -2], [3, 7], [-1, 0]]) \\
 \text{m16} &:= \begin{bmatrix} 2 & 5 \\ 1 & -2 \\ 3 & 7 \\ -1 & 0 \end{bmatrix}
 \end{aligned} \tag{2.225}$$

$$\begin{aligned}
 &> \text{m15} \bullet \text{m16} \\
 &\begin{bmatrix} 44 & 80 \\ 7 & -6 \\ 46 & 67 \end{bmatrix}
 \end{aligned} \tag{2.226}$$

### ***Transposes, Powers, and Equality of Matrices***

Positive powers of matrices in Maple work exactly as you would expect, with the caret symbol, provided, of course, that the matrix is square.

> *m17* := *Matrix*([[1, 2, 3], [4, 5, 6], [7, 8, 9]])

$$m17 := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix} \quad (2.227)$$

> *m17*<sup>5</sup>

$$\begin{bmatrix} 121\,824 & 149\,688 & 177\,552 \\ 275\,886 & 338\,985 & 402\,084 \\ 429\,948 & 528\,282 & 626\,616 \end{bmatrix} \quad (2.228)$$

The transpose of a matrix is most easily computed with the <sup>+</sup> operator, as follows. (We show the 1-D input and the appearance of the input in 2-D mode.)

> *m16*<sup>+</sup>;

$$\begin{bmatrix} 2 & 1 & 3 & -1 \\ 5 & -2 & 7 & 0 \end{bmatrix} \quad (2.229)$$

> *m16*<sup>+</sup>

$$\begin{bmatrix} 2 & 1 & 3 & -1 \\ 5 & -2 & 7 & 0 \end{bmatrix} \quad (2.230)$$

Alternately, you can use the **Transpose** command, but this requires the **LinearAlgebra** package.

> *LinearAlgebra*[*Transpose*](*m16*)

$$\begin{bmatrix} 2 & 1 & 3 & -1 \\ 5 & -2 & 7 & 0 \end{bmatrix} \quad (2.231)$$

Finally, we need to make a note about equality of matrices. For matrices, = does not test that two matrices are equal in the usual sense. For example, observe that the commands below do not yield the expected result.

> *m18* := *Matrix*([[1, 2], [3, 4]])

$$m18 := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix} \quad (2.232)$$

> *m19* := *Matrix*([[1, 2], [3, 4]])

$$m19 := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix} \quad (2.233)$$

> *evalb*(*m18* = *m19*)

$$false \quad (2.234)$$

In fact, what = tests for in the case of matrices is whether or not two names are referring to the same matrix object. This is another case in which being a reference type makes things a bit different.

To test for equality of matrices in the mathematical sense, it is necessary to use the **Equal** command in the **LinearAlgebra** package.

```
> LinearAlgebra[Equal](m18, m19)
true
```

(2.235)

### Zero–One Matrices

With Maple, we can create and manipulate zero–one matrices as well. In particular, we will consider how to compute the meet, join, and Boolean product of zero–one matrices. Unlike matrix addition and multiplication, Maple does not have built-in commands for these computations, so we will need to create our own procedures.

#### Introducing Bits Package Commands

Calculation of meet, join, and the Boolean product require the use of the **and** and **or** bit operations. In the previous chapter we created our own **AND** procedure as a way to explore some fundamental programming constructs. In this section, we instead make use of Maple’s **Bits** package. This package provides several commands related to performing operations on bits and bit strings. We will only make use of two commands: **And** and **Or**. First we load the package.

```
> with(Bits, And, Or) :
```

The **And** and **Or** commands take two arguments and return the bit-wise  $\wedge$  or  $\vee$ , respectively.

```
> And(1, 1)
1
```

(2.236)

```
> And(0, 1)
0
```

(2.237)

```
> Or(0, 1)
1
```

(2.238)

```
> Or(0, 0)
0
```

(2.239)

(Note: the arguments to **And** and **Or** are not restricted to 0 and 1. They can be any integers and Maple will perform the operation on the bit strings that represent the integers. We will not explore that further here as it is not necessary for the task at hand.)

#### A Type for Zero–One Matrices

To create a zero–one matrix, we can use the **Matrix** command as usual.

```
> zol := Matrix([[1, 0, 1], [1, 1, 0], [0, 1, 0]])
```

$$zol := \begin{bmatrix} 1 & 0 & 1 \\ 1 & 1 & 0 \\ 0 & 1 & 0 \end{bmatrix}$$

(2.240)

As has been mentioned before, it is always a good idea to do type-checking in procedures. This is particularly important for the next procedures that we will write. Since we will be using the **And** and **Or** commands from **Bits**, and these commands will *not* produce errors if their inputs are integers other than 0 and 1, it is important for our procedures to make sure that their input are zero–one matrices. Otherwise, the procedures may execute on “bad” inputs and produce nonsense output.

In Maple, a matrix has type **Matrix**, but in this case, we want to be more specific. We will use the following variant type: **'Matrix'(type)**. This allows us to specify the type of the entries that are allowed in the matrix. For example, **'Matrix'(float)** indicates that the entries in the matrix must all be floating point numbers. Since we want to check for zero–one matrices, we use **{0,1}** as the type of entry allowed in the matrix. Remember that braces in a structured type indicate that any of the options inside the braces are acceptable. (Note: the single right quotes are required and delay evaluation so that Maple does not consider the word **Matrix** to be the matrix construction command.)

Observe that **z01** is a zero–one matrix, but **m17** is not.

```
> type(z01, 'Matrix'({0,1}))
true
```

(2.241)

```
> type(m17, 'Matrix'({0,1}))
false
```

(2.242)

### Implementing Join

With those preliminaries completed, we write the join procedure. This procedure will accept two arguments, both of which must be zero–one matrices. The procedure will also need to check the sizes of the matrices. To do this, we use the **RowDimension** and **ColumnDimension** commands from the **LinearAlgebra** package. These commands accept only one argument, the matrix, and return the number of rows or columns, respectively. If the dimensions do not match, an error will be generated.

After confirming that the two matrices are the same size, the procedure will create a new matrix that is the same size as the input matrices. By omitting any initialization information, Maple will automatically fill this matrix with 0s. The procedure will then consider each position in the matrix using two nested for loops to loop through the rows and the columns. For each position, it computes the bitwise **or** of the entries in the original matrices and sets the corresponding entry in the result matrix **R**.

Here is the procedure. Note that we make use of the **uses** statement. This indicates what packages the procedure relies on and ensures that those packages have been loaded, so we can use the short forms of commands.

```
1 BoolJoin := proc(A : 'Matrix'({0,1}), B : 'Matrix'({0,1}))
2   local numrows, numcols, R, r, c;
3   uses LinearAlgebra, Bits;
4   numrows := RowDimension(A);
5   numcols := ColumnDimension(A);
6   if numrows <> RowDimension(B) or numcols <> ColumnDimension(B)
7   then
8     error "Input matrices must be of the same size.";
9   end if;
10  R := Matrix(numrows, numcols, 0);
11  for r from 1 to numrows do
12    for c from 1 to numcols do
13      R[r,c] := Bits[or](A[r,c], B[r,c]);
14    end for;
15  end for;
16  R;
```



```

8   end if ;
9   R := Matrix (numrows, numcols) ;
10  for r from 1 to numrows do
11      for c from 1 to numcols do
12          R[r, c] := Or (A[r, c], B[r, c]) ;
13      end do ;
14  end do ;
15  return R ;
16 end proc:

```

Below we apply this procedure to two example matrices.

$$\begin{aligned}
 &> \text{zo1} \\
 &\quad \begin{bmatrix} 1 & 0 & 1 \\ 1 & 1 & 0 \\ 0 & 1 & 0 \end{bmatrix} \qquad (2.243)
 \end{aligned}$$

$$\begin{aligned}
 &> \text{zo2} := \text{Matrix}([ [1, 0, 0], [1, 1, 1], [0, 0, 0] ]) \\
 &\quad \text{zo2} := \begin{bmatrix} 1 & 0 & 0 \\ 1 & 1 & 1 \\ 0 & 0 & 0 \end{bmatrix} \qquad (2.244)
 \end{aligned}$$

$$\begin{aligned}
 &> \text{BoolJoin}(\text{zo1}, \text{zo2}) \\
 &\quad \begin{bmatrix} 1 & 0 & 1 \\ 1 & 1 & 1 \\ 0 & 1 & 0 \end{bmatrix} \qquad (2.245)
 \end{aligned}$$

We leave the creation of **BoolMeet** to the reader.

### Implementing Boolean Product

We conclude by implementing the Boolean product. Recall two key points from Definition 9 in Section 2.6. First, the size of the product of an  $m \times k$  matrix and an  $k \times n$  is  $m \times n$ , and the product is undefined if the number of columns of the first matrix does not match the number of rows in the second. Second, the  $(i, j)$  entry of the product is given by the formula

$$c_{ij} = (a_{i1} \wedge b_{1j}) \vee (a_{i2} \wedge b_{2j}) \vee \cdots \vee (a_{ik} \wedge b_{kj}).$$

Our Boolean product procedure, **BProduct**, needs to begin by using **RowDimension** and **ColumnDimension** to find the values for  $m$ ,  $k$ , and  $n$  and to raise an error if the number of columns of the first matrix does not match the number of rows of the second. Like **BoolJoin**, we create the result matrix **C** to be of the appropriate size and allow Maple to fill all the entries with 0.

The main work of the procedure is to loop over all the entries of the result matrix and calculate the appropriate value. We use two nested for loops with index variables **i** and **j** representing the rows and columns of the result matrix. Inside these for loops, we need to implement the formula for  $c_{ij}$ . It will be helpful to consider a specific example:

$$(1 \wedge 0) \vee (0 \wedge 0) \vee (0 \wedge 1) \vee (1 \wedge 1) \vee (0 \wedge 1).$$

Note that **And** and **Or** can accept only two arguments, so we cannot use a single **Or** applied to a sequence of **And**s, as you might hope. Instead, we approach this in the following way. First, compute  $1 \wedge 0$ , the first **and**, and store the result as **c**.

```
> c := And(1,0)
   c := 0
```

(2.246)

Then, update **c** to be the result of applying **or** to it and the result of the next **and** term.

```
> c := Or(c,And(0,0))
   c := 0
```

(2.247)

And then repeat with each successive **and**.

```
> c := Or(c,And(0,1))
   c := 0
```

(2.248)

```
> c := Or(c,And(1,1))
   c := 1
```

(2.249)

```
> c := Or(c,And(0,1))
   c := 1
```

(2.250)

In terms of the generic formula, we initialize  $c = (a_{i1} \wedge b_j)$ . Then, we begin a loop with index, say  $p$ , from 2 through  $k$ . At each step in the loop,  $c = c \vee (a_{ip} \wedge b_{pj})$ .

Here is the implementation of **BProduct**.

```
1 BoolProduct := proc(A :: 'Matrix'({0,1}), B :: 'Matrix'({0,1}))
2   local m, k, n, C, i, j, c, p;
3   uses LinearAlgebra, Bits;
4   m := RowDimension(A);
5   k := ColumnDimension(A);
6   if k <> RowDimension(B) then
7     error "Dimension mismatch.";
8   end if;
9   n := ColumnDimension(B);
10  C := Matrix(m, n);
11  for i from 1 to m do
12    for j from 1 to n do
13      c := And(A[i,1], B[1,j]);
```

```

14         for p from 2 to k do
15             c := Or(c, And(A[i, p], B[p, j]));
16         end do;
17         C[i, j] := c;
18     end do;
19 end do;
20 return C;
21 end proc;

```

We test this procedure on the matrices from Example 8 in the textbook.

> *Ex8A* := *Matrix*([[1, 0], [0, 1], [1, 0]])

$$Ex8A := \begin{bmatrix} 1 & 0 \\ 0 & 1 \\ 1 & 0 \end{bmatrix} \quad (2.251)$$

> *Ex8B* := *Matrix*([[1, 1, 0], [0, 1, 1]])

$$Ex8B := \begin{bmatrix} 1 & 1 & 0 \\ 0 & 1 & 1 \end{bmatrix} \quad (2.252)$$

> *BoolProduct*(*Ex8A*, *Ex8B*)

$$\begin{bmatrix} 1 & 1 & 0 \\ 0 & 1 & 1 \\ 1 & 1 & 0 \end{bmatrix} \quad (2.253)$$

## Solutions to Computer Projects and Computations and Explorations

### Computer Projects 3

Given fuzzy sets  $A$  and  $B$ , find  $\bar{A}$ ,  $A \cup B$ , and  $A \cap B$  (see preamble to Exercise 73 of Section 2.2).

*Solution:* In Section 2.2 of this manual, we described Maple's built-in **MultiSet** object, which provides one solution for this problem. That approach would yield little understanding of either fuzzy sets or Maple, however. Instead, we will create a procedure from scratch. We will compute the union and leave complement and intersection to the reader. Recall, from Exercise 74, that the union of fuzzy sets is the fuzzy set in which the degree of membership of an element is the maximum of the degrees of membership of that element in the given sets.

Recall, from the final subsection of Section 2.2 in this manual, that we developed two possible representations of fuzzy sets and procedures to convert between them. We design our procedure to accept the roster representation as input and return a roster representation of the union, since this representation is the most natural for humans to interact with. However, in implementing the union, it is more natural to work with the fuzzy bit string representation of the sets.

Our **FuzzyUnion** procedure will accept as input two fuzzy sets in the roster representation. It proceeds as follows.

1. Determine the effective universe for the two sets: (a) initialize **U** to **{}**; (b) loop over each element of the first set and add the name of that element to **U**; (c) do the same with the second set. Then, **U** will contain the name of all elements appearing in the two sets.
  2. Use **RosterToBit** to convert both sets to their fuzzy-bit representations.
  3. Use the **zip** command with the **max** function on the two fuzzy-bit strings obtained from **RosterToBit**—**zip(max,A,B)** produces the list whose elements are the maximums of the corresponding entries in **A** and **B**.
  4. Use **BitToRoster** on the result to obtain the roster representation.
- Here is the implementation.

```

1 FuzzyUnion := proc (A, B)
2   local U, e, Abits, Bbits, Cbits, C;
3   U := {};
4   for e in A do
5     U := U union {e[1]};
6   end do;
7   for e in B do
8     U := U union {e[1]};
9   end do;
10  Abits := RosterToBit (A, U);
11  Bbits := RosterToBit (B, U);
12  Cbits := zip (max, Abits, Bbits);
13  C := BitToRoster (Cbits, U);
14  return C;
15 end proc;
```

As an example, we will compute the union of the fuzzy sets defined below.

$$\begin{aligned}
 &> \text{fuzzyA} := \{["a", 0.1], ["b", 0.3], ["c", 0.7]\} \\
 &\quad \text{fuzzyA} := \{["a", 0.1], ["b", 0.3], ["c", 0.7]\}
 \end{aligned}
 \tag{2.254}$$

$$\begin{aligned}
 &> \text{fuzzyB} := \{["a", 0.5], ["b", 0.1], ["d", 0.2]\} \\
 &\quad \text{fuzzyB} := \{["a", 0.5], ["b", 0.1], ["d", 0.2]\}
 \end{aligned}
 \tag{2.255}$$

$$\begin{aligned}
 &> \text{FuzzyUnion}(\text{fuzzyA}, \text{fuzzyB}) \\
 &\quad \{["a", 0.5], ["b", 0.3], ["c", 0.7], ["d", 0.2]\}
 \end{aligned}
 \tag{2.256}$$

Procedures for computing intersection and complement are similar and are left as an exercise.

## Computer Projects 9

Given a square matrix, determine whether it is symmetric.

*Solution:* We will create a procedure, **IsSymmetric**, that tests a matrix to see if it is symmetric. Recall that a matrix is symmetric when it is equal to its transpose. Thus, we just need to use the **Equal** command to compare the matrix with the result of applying the **Transpose** command.

We will use the **Matrix** type in the argument to have Maple ensure that only matrices are allowed as arguments.

```

1 | IsSymmetric := proc (M : Matrix)
2 |   LinearAlgebra[Equal] (M, LinearAlgebra[Transpose] (M)) ;
3 | end proc:

```

> *symmetricMatrix* := Matrix([[1, 2, 3], [2, 4, 5], [3, 5, 6]])

$$\text{symmetricMatrix} := \begin{bmatrix} 1 & 2 & 3 \\ 2 & 4 & 5 \\ 3 & 5 & 6 \end{bmatrix} \quad (2.257)$$

> *IsSymmetric* (*symmetricMatrix*)  
*true*

(2.258)

> *notSymmetricMatrix* := Matrix([[1, 2, 3], [4, 5, 6], [7, 8, 9]])

$$\text{notSymmetricMatrix} := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix} \quad (2.259)$$

> *IsSymmetric* (*notSymmetricMatrix*)  
*false*

(2.260)

## Computations and Explorations 2

Given a finite set, list all elements of its power set.

*Solution:* The **powerset** and **subsets** commands were described in Section 2.1 above. We will write a procedure independent of these built-in commands in order to see how such commands might be created.

Recall, from Section 2.2 of the text, that sets may be represented by bit strings. In particular, given a set, say  $\{a, b, c, d, e\}$  for example, a subset may be represented by a string of 0s and 1s provided an order has been imposed on the set. For example, the string 0, 1, 1, 0, 0 corresponds to the subset  $\{b, c\}$ . (Refer to the textbook for a complete explanation.)

In terms of subsets, the bit string representation indicates that, for a given set, there is a one-to-one correspondence between subsets and bit strings. This means that we can solve the problem of listing all subsets of a given set by producing all corresponding bit strings.

To create the bit strings, we follow the approach used in the procedure **NextTA** from Section 1.3 of this manual. Given any bit string, the next string is obtained by working left to right: if a bit is 1, then it gets changed to a 0. When you encounter a 0 bit, it is changed to a 1 and you stop the process. For example, suppose the current string is

1, 1, 1, 0, 0, 1, 0.

You begin on the left changing the first three 1s to 0s. Then, the fourth bit from the left is 0, so this is changed to a 1 and the process stops. The new bit string is

0, 0, 0, 1, 0, 1, 0.

Here is the **NextBitS** (next bit string) procedure. It accepts a bit string and implements the process described above to produce the next bit string. Note that we enforce the type of the input to this procedure using the structured type **list({0,1})** which indicates that the argument to **BitS** must be a list whose elements are 0s and 1s.

```

1 NextBitS := proc (BitS : list({0,1}))
2     local newBitS, i;
3     newBitS := BitS;
4     for i from 1 to nops(newBitS) do
5         if newBitS[i] = 1 then
6             newBitS[i] := 0;
7         else
8             newBitS[i] := 1;
9             return newBitS;
10        end if;
11    end do;
12    return NULL;
13 end proc;
```

> *NextBitS*([1, 1, 1, 0, 0, 1, 0])  
       [0, 0, 0, 1, 0, 1, 0] (2.261)

Next, we will need a way to convert a bit string into a subset of a given set. We can do this using a simplified version of **BitToRoster**. Note that this procedure relies on the fact that Maple imposes an order on sets and that this order is consistent.

```

1 BitToSubset := proc (BitS : list({0,1}), S : set)
2     local subS, i;
3     subS := {};
4     for i from 1 to nops(BitS) do
5         if BitS[i] = 1 then
6             subS := subS union {S[i]};
7         end if;
8     end do;
9     return subS;
10 end proc;
```

> *BitToSubset*([0, 1, 1, 0, 0], {"a", "b", "c", "d", "e"})  
       {"b", "c"} (2.262)

Finally, we combine these two procedures to produce the subsets of a given set.

```

1 Subsets := proc (S : set)
2     local BitS;
```

```

3  BitS := [0 $ nops(S)] ;
4  while BitS <> NULL do
5      print(BitToSubset(BitS, S)) ;
6      BitS := NextBitS(BitS) ;
7  end do ;
8  return NULL ;
9  end proc ;

```

Recall that **0 \$ nops(S)** produces a sequence of **nops(S)** 0s.

We apply our procedure to  $\{a, b, c\}$  to confirm that it is functioning properly.

*> Subsets({“a”, “b”, “c”})*

$\emptyset$

{“a”}

{“b”}

{“a”, “b”}

{“c”}

{“a”, “c”}

{“b”, “c”}

{“a”, “b”, “c”}

(2.263)

## Exercises

**Exercise 1.** Write a procedure **AreDisjoint** that accepts two sets as arguments and returns true if the sets are disjoint and false otherwise.

**Exercise 2.** Write a procedure, **Cartesian**, to compute the Cartesian product of two sets as a single set.

**Exercise 3.** Write procedures **FuzzyIntersection** and **FuzzyComplement** to complete Computer Project 3.

**Exercise 4.** Write procedures for computing the complement, union, intersection, difference, and sum for multisets, without using the built-in **MultiSet** object type. Instead, represent a multiset as a set of pairs **[a,m]** where **m** is the multiplicity of the element **a**. (Refer to Section 2.2 for information about multisets.)

**Exercise 5.** Write procedures to compute the Jaccard similarity and Jaccard distance between sets. (Refer to the preamble of Exercise 71 in Section 2.2 for the definitions of Jaccard similarity and distance.)

**Exercise 6.** Write procedures to compute the image of a finite set under a function. Create one procedure for functions defined as a procedure or a functional operator and a second procedure for functions defined via tables.

**Exercise 7.** Write a procedure to find the inverse of a function defined by a table.

**Exercise 8.** Write a procedure to find the composition of functions defined by tables.

**Exercise 9.** Use computation to discover what the largest value of  $n$  is for which  $n!$  has fewer than 1000 digits. (Hint: the **length** command applied to an integer will return the number of digits of the integer.)

**Exercise 10.** Write a procedure **ArithmeticSeq**, similar to **GeometricSeq** from above, that produces an arithmetic sequence.

**Exercise 11.** Find the first 20 terms of the sequences defined by the recurrence relations below.

- a)  $a_n = 2a_{n-1} + 3a_{n-2}$ , with  $a_1 = 1$  and  $a_2 = 0$ .
- b)  $a_n = a_{n-1} + na_{n-2} + n^2a_{n-3}$ , with  $a_1 = 1$ ,  $a_2 = 1$ , and  $a_3 = 3$ .
- c)  $a_n = a_{n-1}a_{n-2} + 1$ , with  $a_1 = a_2 = 1$ .

**Exercise 12.** The Lucas numbers satisfy the recurrence  $L_n = L_{n-1} + L_{n-2}$  and the initial conditions  $L_1 = 2$  and  $L_2 = 1$ . Use Maple to gain evidence for conjectures about the divisibility of Lucas numbers by different integer divisors.

**Exercise 13.** Write a procedure to find the first  $n$  Ulam numbers and use the procedure to find as many Ulam numbers as you can. (Ulam numbers are defined in Exercise 28 of the Supplemental Exercises for Chapter 2.)

**Exercise 14.** Use Maple to find formulae for the sum of the  $n$ th powers of the first  $k$  positive integers for  $n$  up to 10.

**Exercise 15.** The calculation of **actualValues** above is very inefficient, because Maple must calculate the entire list of rational numbers for each value from 2 to 100. Create a new procedure, **listActuals**, that accepts a maximum stage as input and returns the list whose entries are the values of  $F(n)$ . You can do this by modifying **ListRationals** so that at the completion of each stage, the size of **L**, that is, the value  $F(n)$ , is recorded in a list.

**Exercise 16.** Find a value  $R$  so that the graph of  $R \cdot \frac{n(n-1)}{2}$  is just above the graph of  $F(n)$ . Use your **listActuals** procedure to expand the data and refine the value of  $R$ .

**Exercise 17.** Use Maple to find the 100th positive rational number in the list generated by **ListRationals**. What about the 1000th? 10 000th? (If you completed it, the result of the previous exercise can be helpful.)

**Exercise 18.** Write a procedure **BoolMeet** implementing the Boolean meet operation on zero-one matrices.